

## 17 The Tribon Geometry Macro Facility

## 17.1 General

The Tribon Geometry Macro Facility is a program used to create predefined geometry. The main purpose of using geometry macros in Tribon M3 is to create design standards. Example of such standards are profile cutouts in Tribon Hull and ventilation volumes in Tribon Ventilation.

**Note:** Previous versions of Tribon further allowed any user to define geometry macros creating drawings and sub-pictures. However, this is in Tribon M3 no longer supported. Instead Tribon Vitesse functions for Drafting should be used.

The geometry is put into the current drawing or volume.

## 17.2 Introduction

In interactive Tribon applications, 2D drawings are built up by different interactive functions of the applications. Similarly, 3D figures can be created by the combination of volume primitives. These operations normally work on entity level, i.e. by the addition of lines, arcs, texts, etc. or by the duplication of information of the drawing.

However, in many situations the drawing or volume to be created is parenthesized, i.e. controlled by a relatively small number of parameters and/or conditions.

The Tribon Geometry Macro Facility has been developed as a tool to create such drawings/volumes by defining a number of parameters in calls of **geometry macros**.

A geometry macro is written as a text in a format like a programming language.

The elements of the language are:

- | the geometrical entities that can be created
- | "program logic", like branching and loops
- | basic arithmetics, logical and trigonometrical operations

Details about all these things can be found below.

## 17.3 Functions

## 17.3.1 Entities of the Geometry Macro

A geometry macro used for a drawing or a sub-picture may generate the following entities:

- | point
- | vector
- | line
- | arc
- | circle
- | contour
- | spline
- | symbol
- | text
- | text file
- | layer
- | line type
- | attribute
- | note
- | hatch pattern

A geometry macro used for volumes may generate the following entities:

- | point
- | vector
- | parallelepiped
- | cylinder
- | cone
- | spherical segment
- | general cylinder
- | toroid
- | polygon
- | rotational primitive
- | attribute

One geometry macro can define either a 2D drawing/sub-picture or a 3D volume. However, some of the volume primitives use 2D entities as parameters.

## 17.3.2 Additional Functions

The following functions can be used in any macro:

- | assign
- | colour
- | branching: if..
  - .
  - .
  - .
  - else
  - .
  - .
  - .
  - endif
- | loop
- | conditional loop: while..
  - .
  - .
  - .
  - endwhile
- | table (formatted output)

## 17.3.3 Operators and System Functions

<b>AND</b>	logical and
<b>+</b>	add
<b>-</b>	change sign
<b>&amp;</b>	concatenate strings
<b>%</b>	division
<b>**</b>	exponentiation
<b>==</b>	qual
<b>&gt;</b>	greater than
	greater equal
<b>&lt;</b>	less than
	less equal
<b>*</b>	multiplication
<b>NOT</b>	logical not
<b>...</b>	not equal
<b>OR</b>	logical or
<b>-</b>	subtraction
<b>XOR</b>	logical exclusive or
<b>AB</b>	absolute value
<b>ACO</b>	arcus cosine (radians)
<b>ACOSD</b>	arcus cosine (degrees)
<b>ASIN</b>	arcus sine (radians)
<b>ASIND</b>	arcus sine (degrees)
<b>ATAND</b>	arcus tangent (radians)
<b>ATAND</b>	arcus tangent (degrees)
<b>BYTE</b>	character whose ASCII code is the argument of BYTE
<b>COS</b>	cosine (radians)
<b>COSD</b>	cosine (degrees)
<b>DEGREE</b>	angle in degrees to angle in radians
<b>SIN</b>	sine (radians)
<b>SIND</b>	sine (degrees)
<b>SQRT</b>	square root
<b>TAN</b>	tangent (radians)
<b>TAND</b>	tangent (degrees)
<b>SUBSTR</b>	substrings
<b>LENGTH</b>	length of a string
<b>INDEX</b>	position of a specified sub-string within a string
<b>NRCHAR</b>	number of character with specified number of decimals

## 17.3.4 Creation of a Geometry Macro

The geometry macro is created by writing a geometry macro source text in a special macro language. This language is similar to other languages used for input to various Tribon applications. The syntax of the Tribon Geometry Macro Language is given in [Syntax of Tribon Geometry Macro Language](#). The source text is stored in an ordinary text file with an arbitrary name (= the name of the macro). It can thus be created using an ordinary editor. Only capital letters, digits and \_ (underscore) are allowed characters in the macro and file name.



## 17.3.5 Execution of a Geometry Macro

The geometry macro can be executed either from Tribon Drafting or as a stand alone program. The following is valid for both cases.

The first time the macro is to be executed, the file name must be given, including the file type. It is assumed that the geometry macro is stored on the directory given by the logical variable SBB\_GEO\_MACRO\_SRC. The interpreter will then check the syntax of the macro in the source file (the macro is "compiled").

The result of the interpretation is given in a list file containing a list of the syntax together with possible error messages. The name of the list file is `<source_file>.LST` and the list file will be created on the directory given by the logical variable SBB\_GEO\_MACRO\_LST.

If the interpretation was successful, a "compiled" version of the syntax will be stored on a directory given by the logical variable SBB\_GEO\_MACRO\_BIN. The name of the "compiled" version of the syntax is `<source_file>.GLB`. At the subsequent executions of the geometry macro, it is sufficient to give the macro name and the "compiled" version will be read, after which the execution will start.

Note, the following paragraph is only valid when a project is shared between Windows and UNIX/VMS.

In case that the Tribon Geometry Macro is used in a project that is shared between platforms, and the project is located on a UNIX or VMS machine, the file handling differs a bit from what is described above. The list file emanating from a compilation on Windows (`<source_file>.LST`) will be placed in the directory indicated by the operating system environment variable TEMP, e.g. c:\Temp. Since the compiled format differs between platforms, it is necessary to have the compiled files (`<source_file>.GLB`) stored on each platform respectively. This also means that the macros have to be compiled once on each platform. The logical variable SBB\_GEO\_MACRO\_BIN\_NT is used to indicate the directory used for compiled binary files on the Windows platform. Note that the source files are still fetched from the directory indicated by SBB\_GEO\_MACRO\_SRC, even if that is on another platform. If you do not share projects, that is, you run only on the Windows platform, you do not have to set the variable SBB\_GEO\_MACRO\_BIN\_NT, and the compiled files will end up in the directory SBB\_GEO\_MACRO\_BIN.

## Stand Alone Program

If the Geometry Macro is run as a stand alone program, the result can be presented in a number of different ways. The following activities are available:

1. Print on terminal
2. Create 2D geometry and store on data bank
3. Create 3D volume model and store on data bank
4. Create 3D volume model + picture and store on data bank

## 17.3.6 Parameters of a Geometry Macro

The geometry macro can be parameterised, i.e. some parameters are given values at the time of execution. A parameter can be an integer number, a decimal number, a text string or a 2D or 3D position in the drawing/volume. It is possible to define a command string to each parameter. This string will be displayed at the workstation and the operator is prompted to give the parameter value.

## Submacros in a Geometry Macro

It is possible to call another geometry macro (submacro) from any macro. Data is transferred from the macro to the submacro by parameters. These parameters can have any type.

The submacro parameters will be assigned the values of the corresponding macro parameters and the submacro will be executed. When the submacro execution is finished the macro parameters will be assigned the values of the corresponding submacro parameters and the execution of the macro is continued. See also [The Facility of Using Conditional Statements in a Macro](#).

The macro and the submacro can be the same macro so the resulting macro is recursive. See also [The Possibility to Write Recursive Geometry Macros](#).

## 17.3.7 Error Handling

During the execution of a geometry macro, an error handling system is active so that abortions of Tribon Drafting due to macro programming errors will be prevented.

The programmer has the possibility to list any data at the workstation during execution. This will make it easier to detect any errors in the macro programming.

## 17.3.8 Execution in Batch

It is possible to execute a geometry macro in batch. All input to the macro must then be available in an ordinary text file and in exactly the same order as they are required by the macro.

The first parameter in the file must then be the activity and the second one the name of the macro.

## 17.4 Appendices

## 17.4.1 Syntax of Tribon Geometry Macro Language



## Conventions Used in this Document

In this document the following conventions are used (cf. *"The Tribon Interpretative Language"*):

[ ] optional

< > term

....preceding expression may be repeated

## Statement Types

The input language contains the following different statement types:

ARC	The ARC statement defines a 2D arc. The arc can be used as input to the PRESENT and CONTOUR statements.
ASSIGN	The ASSIGN statement assigns a previously defined variable to another variable which will get the same type as the first one. However, if a 2D or a 3D point has been defined, one of these coordinates can be assigned to a variable with type decimal.
ATTRIBUTE	The ATTRIBUTE statement is used to put attributes in the object. The attributes contain non-geometrical data and it is possible to store 100 integers, 50 reals and 12 strings. Both attribute numbers and the variables can be addressed with an alias name defined in an alias file assigned to the logical variable SBD_ALIAS.
CALL	The CALL statement is used to call another macro. The parameter types must match and the parameters can be used for either input or output purposes. There is no limit for the number of levels of submacros.
CHANGEDRAW	The CHANGEDRAW statement is used to change the appearance of an object, a component or a subcomponent. This can be done either in one view or in all views.
CIRCLE	The CIRCLE statement defines a 2D circle. The circle can be used as input to the PRESENT statement.
COLOUR	The COLOUR statement sets the modal colour. Default colour is green. If the given colour is an empty string, the background colour will be used.
CONE	The CONE statement defines a 3D cone. The cone can be used as input to the PRESENT statement.
CONNECTIONPOINT	The CONNECTIONPOINT statement defines a 3D connection point. The connection point can be used as input to the PRESENT statement.
CONTOUR	The CONTOUR statement defines a 2D contour. The contour can be used as input to the GENERALCYLINDER and PRESENT statements.
CURRENT	The CURRENT statement makes it possible to structure a drawing or a volume.
CYLINDER	The CYLINDER statement defines a 3D cylinder. The cylinder can be used as input to the PRESENT statement.
DECLARE	The DECLARE statement allows the user to specify the type of any variable. The default type is DECIMAL. The DECLARE statements must be given directly after the MACRO statement.
DELAY	The DELAY statement makes it possible to introduce a delay in a macro. It has only an effect when geometry is presented at a workstation.
DISTANCE	The DISTANCE statement assigns a value or a previously defined variable to another variable. The value is scaled with the current scale.
DRAWING_NAME	The DRAWING_NAME statement assigns the name of the current drawing to it's parameter.
ELSE	The ELSE statement must be used together with the IF statement. If the condition given in the IF statement is not fulfilled, then the statements after the ELSE statement will be executed.
ENDIF	The ENDIF statement terminates the IF statement.
ENDLOOP	The ENDLOOP statement terminates the LOOP statement.
ENDMACRO	The ENDMACRO statement terminates the MACRO statement.
ENDWHILE	The ENDWHILE statement terminates the WHILE statement.
EXTRACTION	The EXTRACTION statement makes it possible to use Data Extraction to get any data item of a stored Tribon object. This statement is optional.

GENERALCYLINDER	The GENERALCYLINDER statement defines a 3D general cylinder. The general cylinder can be used as input to the PRESENT statement.
GET	The GET statement enables the user to get a number of variables which will be used as parameters.
HATCH	The HATCH statement defines a hatch pattern in a drawing. User defined patterns can be created. The hatch pattern may include islands. The hatch pattern is used as input to the PRESENT statement.
IF	The IF statement tests an expression and performs a specified action if the result of the test is true. All statements between the IF and ENDIF statements will be executed. If the expression is false, no action will be taken unless the ELSE statement is given. In this case, all statements between the ELSE and ENDIF statements will be executed. All statements except the MACRO and ENDMACRO statements can be given after the IF statement. The IF statements can be nested.
LAYER	The LAYER statement sets the modal layer. The default layer is 0. The layer can be given either as a number or as an alias if an alias file exists. The aliases should be defined in a file assigned to the logical variable SB_LAYER_ALIAS. In this case, geometry, text and symbols will get the same modal layer. To give them different layer values, it is possible to use layer classes. The classes are defined in a file assigned to the logical variable SBD_LAYER_CLASS.
LINE	The LINE statement defines a 2D line. The line can be used as input to the CONTOUR and PRESENT statements.
LINETYPE	The LINETYPE statement sets the modal line type. Default line type is solid. The width can also be changed. Default line width is thin.
LOOP	The LOOP statement makes it possible to execute the same statements a number of times. All statements except the MACRO and ENDMACRO statements can be given after the LOOP statement. The LOOP statements can be nested.
MACRO	The MACRO statement must be the first statement in the macro. Parameters may be given and, in that case, they will be given values before the execution of the macro.
NAME	The NAME statement is used to define a name on the drawing/ subpicture/volume to be created and stored on a data bank. If the statement is omitted, the macro name will be used. It is also possible to give the form to be used, if any. If the form is omitted, a drawing will be created with no form. The scale can also be given for drawings and subpictures.
NOTE	The NOTE statement defines a note in a drawing. The note is used as input to the PRESENT statement.
PARALLELEPIPED	The PARALLELEPIPED statement defines a 3D parallelepiped. The parallelepiped can be used as input to the PRESENT statement.
POINT_2D	The POINT_2D statement defines a 2D point. The point is used as input to several of the other 2D statements. It is also possible to use this statement when any of the coordinates are to be changed.
POINT_3D	The POINT_3D statement defines a 3D point. The point is used as input to several of the other 3D statements. It is also possible to use this statement when any of the coordinates are to be changed.
POLYGON	The POLYGON statement defines a 3D polygon. The polygon is used as input to the PRESENT statement.
PRESENT	The PRESENT statement presents the geometry created by the macro.
PUT	The PUT statement makes it possible to print any variable data.
RANGE	The RANGE statement defines a range of variables. The range is used in the EXTRACTION and the LOOP statements.
ROTATIONAL	The ROTATIONAL statement defines a 3D rotational primitive. The rotational primitive is used as input to the PRESENT statement.
SPHERESEG	The SPHERESEG statement defines a 3D spherical segment. The segment is used as input to the PRESENT statement.
SPLINE	The SPLINE statement defines a 2D spline. The spline is used as input to the

	CONTOUR and PRESENT statements.
SPLIT	The SPLIT statement splits a model name into project, module and subsystems. These items can then be used in the EXTRACT statement.
SYMBOL	The SYMBOL statement defines a symbol. The symbol is used as input to the PRESENT statement.
TABLE	The TABLE statement makes it possible to present a number of variables with a format determined by the user. The table is used in the PRESENT statement.
TEXT	The TEXT statement defines a text. The text is used as input to the PRESENT statement.
TEXTFILE	The TEXTFILE statement defines a text file. The text file is used as input to the PRESENT statement.
TOROID	The TOROID statement defines a 3D toroid. The toroid is used as input to the PRESENT statement.
VECTOR_2D	The VECTOR_2D statement defines a 2D vector. The vector is used as input to other 2D statements. It is also possible to use this statement when any of the coordinates shall be changed.
VECTOR_3D	The VECTOR_3D statement defines a 3D vector. The vector is used as input to several of the other 3D statements. It is also possible to use this statement when any of the coordinates are to be changed.
VMSJOB	The VMSJOB statement is used to execute a job code or command VMS-file.
WHILE	The WHILE statement makes it possible to execute the same statements a number of times. The WHILE statement tests an expression and performs a specified action as long as the result of the test is true. All statements between the WHILE and ENDWHILE statements will be executed. If the expression is false, no action will be taken. All statements except the MACRO and ENDMACRO statements can be given after the WHILE statement. The WHILE statements can be nested.

## Statement Syntax

Below, the complete syntax of each statement type is described.

### The ARC Statement

```
ARC,<arc_name>,<start_pnt>
[ /ARCMIDPNT=( <mid_pnt>,<end_pnt> ) ]
[ /ARCRADIUS=( <end_pnt>,<rad> ) ]
[ /ARCAMPLITUDE=( <end_pnt>,<ampl> ) ] ;
```

<arc\_name> is the name of the arc and will be assigned the type ARC\_2D. The maximum length of <arc\_name> is 32 characters.

<start\_pnt> is the starting point of the arc with type POINT\_2D.

#### **ARCMIDPNT=(<mid\_pnt>,<end\_pnt>)**

<mid\_pnt> is the mid point of the arc when the arc is defined by giving three points. It has the type POINT\_2D.

<end\_pnt> is the ending point of the arc with type POINT\_2D.

#### **ARCRADIUS=(<end\_pnt>,<rad>)**

<end\_pnt> is the ending point of the arc with type POINT\_2D.

<rad> is the arc radius when the arc is defined by giving two points + radius. It has the type DECIMAL.

#### **ARCAMPLITUDE=(<end\_pnt>,<ampl>)**

<end\_pnt> is the ending point of the arc with type POINT\_2D.

<ampl> is the arc amplitude when the arc is defined by giving two points + amplitude. It has the type DECIMAL.

### Structure:

```
NSEG                                (INTEGER)
SEGPARTS ( 1:NSEG )
                                ENDPNT ( 1:2 )    (DECIMAL)
                                AMPLITUDE ( 1:2 ) (DECIMAL)
```

#### **Example:**

```
GET/STRUCTURE=(N,<arc_name>,'NSEG')
/STRUCTURE=(X,<arc_name>,'ENDPNT',N,1)
/STRUCTURE=(Y,<arc_name>,'AMPLITUDE',2,'Y');
```

### The ASSIGN Statement

```
ASSIGN,<variable_1>,<variable_2>
```

```
[/XCOORD]
```

```
[/YCOORD]
```

```
[/ZCOORD];
```

<variable\_1> is assigned the same value as <variable\_2> and will get the same type as the first one. The maximum length of <variable\_1> and <variable\_2> is 32 characters.

<variable\_2> can be an expression or have any of the following types:

INTEGER

DECIMAL

STRING

POINT\_2D  
LINE\_2D  
ARC\_2D  
CONTOUR\_2D  
CIRCLE\_2D  
VECTOR\_2D  
SPLINE\_2D  
TEXTFILE\_2D  
TEXT\_2D  
SYMBOL\_2D  
NOTE\_2D  
HATCH\_2D  
EXTRACT  
RANGE  
TABLE  
POINT\_3D  
VECTOR\_3D  
CONNECTIONPOINT\_3D  
CONE\_3D  
CYLINDER\_3D  
GENERALCYLINDER\_3D  
PARALLELEPIPED\_3D  
POLYGON\_3D  
SPHERESEG\_3D  
TOROID\_3D  
ROTATIONAL\_3D

#### **XCOORD**

If <variable2> is of type POINT\_2D or POINT\_3D, then <variable1> can be assigned the x coordinate of <variable2>. This variable will get type DECIMAL.

#### **YCOORD**

If <variable2> is of type POINT\_2D or POINT\_3D, then <variable1> can be assigned the y coordinate of <variable2>. This variable will get type DECIMAL.

#### **ZCOORD**

If <variable2> is of type POINT\_3D, then <variable1> can be assigned the z coordinate of <variable2>. This variable will get type DECIMAL.

### **The ATTRIBUTE Statement**

**ATTRIBUTE,<attr\_no>**

**/ATTRDATA=(<variable>,<attribute>)....;**

<attr\_no> is the attribute number (> 0) and must be given within ' ' if it is addressed with an alias name. If <attr\_no> is a true number or a variable, it shall not be given within ' '. The maximum length of <attr\_no> as an alias is 26 characters and as a variable 32 characters.

**ATTRDATA=(<variable>,<attribute>)....**

<variable> is the variable name given as an alias or using the default names. These are I1 - I100 for the integers, R1 - R50 for the reals and S1 - S12 for the strings. <variable> shall always be given within ' '. The maximum length of <variable> as an alias is 26 characters and as a variable 32 characters.

<attribute> is the data to be stored in the attribute as an integer, real or string (max 26 characters). <attribute> must be given within ' ' if it is a string but not if it is a true number or a variable. The maximum length of <attribute> as a variable is 32 characters.

## The CALL Statement

**CALL,<macro\_name>**

**[,<arg\_1>[,<arg\_2>...[,<arg\_25>]...]];**

<macro\_name> is the name of the submacro. The maximum length of <macro\_name> is 32 characters. Only upper case letters and \_ are allowed characters. <arg\_1>,<arg\_2>,...,<arg\_25> are the arguments to the submacro. They cannot be expressions but must be variables. Any type is allowed. It is important to notice that if a parameter in the submacro is changed, the corresponding argument in the calling macro will also be changed. Thus, it is of great advantage if an argument is used for either input or output.

## The CHANGEDRAW Statement

**CHANGEDRAW,<obj\_type>,<obj\_name>**

**/COMPID=(<comp\_id>)**

**/MARKINGCOLOUR=(<marking\_colour>)**

**/SUBCOMPID=(<subcomp\_id>)**

**/VIEW=(<view\_id>);**

<obj\_type> is a keyword describing the type of the object. The following keywords exist:

**PANEL** plane and curved panel objects

**CWAY** cableway object

**CABLE** cable objects

**EQUIP** equipment object

The object type must be a STRING constant.

<obj\_name> is the name of the object to be changed, and has the type STRING.

<comp\_id> is the id of a component of the type INTEGER, and is only used in the case of changing a single component.

<marking\_colour> is the new colour of the object. It has the type STRING. The maximum length of <marking\_colour> is 32 characters. If DEFAULT is assigned as <marking\_colour>, the geometry will be redrawn in the default colour. In a modelling view the default colour is determined by the model and in a diagram view the default colour is determined by the General Diagram default file and, if applicable, pipe specification.

See *Tribon M3 Drafting Appendices* for valid Tribon colours

<subcomp\_id> is used in the same manner as <comp\_id>, but for subcomponents. It has the type INTEGER.

<view\_id> is the id of the view in which to change the object, and has the type INTEGER. If not given, the appearance of the object is changed in all views of the current drawing.

## The CIRCLE Statement

**CIRCLE,<circ\_name>,<circ\_cent>,<circ\_rad>;**

<circ\_name> is the name of the arc and will be assigned the type CIRCLE\_2D. The maximum length of <circ\_name> is 32 characters.

<circ\_cent> is the centre point of the circle with type POINT\_2D.

<circ\_rad> is the radius of the circle. It has the type DECIMAL.

### Structure:

NSEG (INTEGER)

SEGPARTS (1:NSEG)

ENDPNT (1:2) (DECIMAL)

AMPLITUDE (1:2) (DECIMAL)

:

**Example:**

```
GET/STRUCTURE=(N,<circ_name>,'NSEG')
/STRUCTURE=(X,<circ_name>,'ENDPNT',N,1)
/STRUCTURE=(Y,<circ_name>,'AMPLITUDE',2,'Y');
```

### The COLOUR Statement

**COLOUR,<col>;**

<col> is the new modal colour. It has the type STRING. The maximum length of <col> is 32 characters.

See *Tribon M3 Drafting Appendices* for valid Tribon colours.

Empty String means the background colour.

### The CONE Statement

**CONE,<cone\_name>,<rad\_1>,<rad\_2>**

**/COORDCONE=(<cl\_pnt\_1>,<cl\_pnt\_2>);**

<cone\_name> is the name of the cone and will be assigned the type CONE\_3D. The maximum length of <cone\_name> is 32 characters.

<rad\_1> and <rad\_2> are the radii of the bottom and top circles respectively with type DECIMAL.

**COORDCONE=(<cl\_pnt\_1>,<cl\_pnt\_2>)**

<cl\_pnt\_1> and <cl\_pnt\_2> are the centre line starting and ending points, respectively. They have the type POINT\_3D.

#### Structure:

PNT(1:3) (POINT\_3D)  
 VEC(1:3) (VECTOR\_3D)  
 BASE (DECIMAL)  
 TOP (DECIMAL)

#### Example:

```
GET/STRUCTURE=(X,<cone_name>,'PNT',1)
/STRUCTURE=(Z,<cone_name>,'VEC',3)
/STRUCTURE=(R1,<cone_name>,'BASE')
/STRUCTURE=(R2,<cone_name>,'TOP');
```

### The CONNECTIONPOINT Statement

**CONNECTIONPOINT,<conn\_name>,<conn\_type>,<conn\_no>,<conn\_pnt>,<conn\_vect>,<conn\_desc>;**

<conn\_name> is the name of the connection point and will be assigned the type CONNECTIONPOINT\_3D. The maximum length of <conn\_name> is 32 characters.

<conn\_type> is the connection type of type INTEGER. Only types between 1 and 9 are valid.

<conn\_no> is the connection number of type INTEGER. Only numbers between 1 and 199 are valid.

<conn\_pnt> is the connection point defining the position. It has the type POINT\_3D.

<conn\_vect> is the connection vector defining the direction. It has the type VECTOR\_3D.

<conn\_desc> is the connection description of type STRING. Maximum length of <conn\_desc> is 100 characters.

#### Structure:



CONTYPE	(INTEGER)
CONNUMBER	(INTEGER)
PNT(1:3)	(DECIMAL)
VEC(1:3)	(DECIMAL)
DESCR	(STRING)

#### Example:

```
GET/STRUCTURE=(T,<conn_name>,'CONTYPE')
/STRUCTURE=(N,<conn_name>,'CONNUMBER')
/STRUCTURE=(X,<conn_name>,'PNT',1)
/STRUCTURE=(Y,<conn_name>,'VEC','Z')
/STRUCTURE=(D,<conn_name>,'DESCR');
```

### The CONTOUR Statement

```
CONTOUR,<cnt_name>[,<start_pnt>]
[/ARC=<arc_name>]
[/ARCMIDPNT=(<mid_pnt>,<end_pnt>)]
[/ARCRADIUS=(<end_pnt>,<rad>)]
[/ARCAMPLITUDE=(<end_pnt>,<ampl>)]
[/CONTOUR=<cnt_name>]
[/LINE=<line_name>]
[/LINEEND=<end_pnt>]
[/LINEANGLE=(<len>,<ang>)]
[/LINEOFFS=<offs>]
[/SPLINE=<spl_name>];
```

<cnt\_name> is the name of the contour and will be assigned the type CONTOUR\_2D. The maximum length of <cnt\_name> is 32 characters.

<start\_pnt> is the starting point of the contour with type POINT\_2D.

If /ARC, /CONTOUR, /LINE or /SPLINE is used for the first segment of the contour, then <start\_pnt> shall be omitted.

#### **ARC=<arc\_name>**

<arc\_name> is the name of the arc and has the type ARC\_2D. If <arc\_name> is not the first segment in the contour, the starting point of <arc\_name> will be ignored. No check is made whether this point coincides with the ending point of the previous segment or not.

#### **ARCMIDPNT=(<mid\_pnt>,<end\_pnt>)**

<mid\_pnt> and <end\_pnt> are the mid and ending points of the arc. They have the type POINT\_2D. The starting point is defined by <start\_pnt> or by the ending point of the previous segment.

#### **ARCRADIUS=(<end\_pnt>,<rad>)**

<end\_pnt> is the ending point of the arc with the type POINT\_2D. The starting point is defined by <start\_pnt> or by the ending point of the previous segment.

<rad> is the arc radius when the arc is defined by giving two points + radius. It has the type DECIMAL.

#### **ARCAMPLITUDE=(<end\_pnt>,<ampl>)**

<end\_pnt> is the ending point of the arc with the type POINT\_2D. The starting point is defined by <start\_pnt> or by the ending point of the previous segment.

<ampl> is the arc amplitude when the arc is defined by giving two points + amplitude. It has the type DECIMAL.

#### **CONTOUR=<cnt\_name>**

<cnt\_name> is the name of the contour and has the type CONTOUR\_2D. If <cnt\_name> is not

the first segment in the contour, the starting point of <cnt\_name> will be ignored. No check is made whether this point coincides with the ending point of the previous segment.

**LINE=<line\_name>**

<line\_name> is the name of the line and has the type LINE\_2D. If <line\_name> is not the first segment in the contour the starting point of <line\_name> will be ignored. No check is made whether this point coincides with the ending point of the previous segment or not.

**LINEEND=<end\_pnt>**

<end\_pnt> is the ending point of the line with the type POINT\_2D. The starting point is defined by <start\_pnt> or by the ending point of the previous segment.

**LINEANGLE=(<len>,<ang>)**

<len> is the length of the line and <ang> is the angle of the line, both with type DECIMAL. The starting point is defined by <start\_pnt> or by the ending point of the previous segment.

**LINEOFFS=<offs>**

<offs> is the offset from the current point. It has the type STRING with at most 72 characters. The following formats are valid:

du, dv (e.g. 100,50) or

length, angle (e.g. 100,45D) or

length, verbal direction (e.g. 100,N).

Verbal directions can be North, South, West, East, Right, Left, Up or Down. Only the first letter is relevant. If the string ends with a 'U' or a 'V', the u axis or the v axis will be locked. Angles must be followed by D (degrees) or R (radians).

The strings must only contain digits, a '-' or a '.'. '-' is only valid in the first position. Between the strings, it is allowed to have a couple of ' ' and ',' but at least one, either a ' ' or a ','. If one of the letters mentioned above is at a valid position, the rest of the string is ignored. Spaces in the beginning or the end of the string are removed.

**SPLINE=<spl\_name>**

<spl\_name> is the name of the spline and has the type SPLINE\_2D. If <spl\_name> is not the first segment in the contour, the starting point of <spl\_name> will be ignored. No check is made whether this point coincides with the ending point of the previous segment or not.

**Structure:**

NSEG	(INTEGER)
SEGPARTS ( 1 : NSEG )	
ENDPNT ( 1 : 2 )	(DECIMAL)
AMPLITUDE ( 1 : 2 )	(DECIMAL)

**Example:**

```
GET/STRUCTURE=(N,<cnt_name>,'NSEG')
/STRUCTURE=(X,<cnt_name>,'ENDPNT',N,1)
/STRUCTURE=(Y,<cnt_name>,'AMPLITUDE',2,'Y');
```

**The CURRENT Statement**

**CURRENT[/SUBPICTURE=(<view\_name>,<subview\_name>)**

**[/VIEWSCALE=<view\_scl>]]**

**[/SUBVOLUME=<subvol\_no>];**

**SUBPICTURE=(<view\_name>,<subview\_name>)**

**[/VIEWSCALE=<view\_scl>]**

<view\_name> is the name of the view. It has the type STRING. The maximum length of <view\_name> is 26 characters. If <view\_name> does not exist a new view and a new subview are created. The subview will get the name <subview\_name>.

<subview\_name> is the name of the subview. It has the type STRING. The maximum length of <subview\_name> is 26 characters. If <subview\_name> does not exist a new subview is created.

<view\_scl> is the scale to be used when a new view is created. It has the type DECIMAL. If this attribute is not given then the view scale will be the same as the drawing scale (given in the NAME statement).

A new component is always created when a macro is run. This is the case even if the CURRENT statement is not present.

**SUBVOLUME=<subvol\_no>**

<subvol\_no> is the number of the subvolume. It has the type INTEGER. If <subvol\_no> does not exist it is created.

## The CYLINDER Statement

**CYLINDER,<cyl\_name>,<rad>**

**/COORDCYL=(<cl\_pnt\_1>,<cl\_pnt\_2>;**

<cyl\_name> is the name of the cylinder and will be assigned the type CYLINDER\_3D. The maximum length of <cyl\_name> is 32 characters.

<rad> is the radius of the cylinder with type DECIMAL.

**COORDCYL=(<cl\_pnt\_1>,<cl\_pnt\_2>)**

<cl\_pnt\_1> and <cl\_pnt\_2> are the centre line starting and ending points, respectively. They have the type POINT\_3D.

### Structure:

PNT(1:3)	(DECIMAL)
VEC(1:3)	(DECIMAL)
BASE	(DECIMAL)

### Example:

GET/STRUCTURE=(X,<cyl_name>,'PNT',1) /STRUCTURE=(Y,<cyl_name>,'VEC','Y') /STRUCTURE=(R,<cyl_name>,'BASE');
--

## The DECLARE Statement

**DECLARE,<variable>,<type>;**

<variable> is the name of the variable whose type is to be declared. The maximum length of <variable> is 32 characters.

<type> is the type of <variable>. It has the type STRING and the whole type name must be given. The following types are available:

INTEGER

DECIMAL

STRING

POINT\_2D

LINE\_2D

ARC\_2D

CONTOUR\_2D

CIRCLE\_2D

VECTOR\_2D

SPLINE\_2D

TEXTFILE\_2D

TEXT\_2D

SYMBOL\_2D

NOTE\_2D

HATCH\_2D

EXTRACT  
RANGE  
TABLE  
POINT\_3D  
VECTOR\_3D  
CONNECTIONPOINT\_3D  
CONE\_3D  
CYLINDER\_3D  
GENERALCYLINDER\_3D  
PARALLELEPIPED\_3D  
POLYGON\_3D  
SPHERESEG\_3D  
TOROID\_3D  
ROTATIONAL\_3D  
The default type is DECIMAL.

### ***The DELAY Statement***

**DELAY,<del\_time>;**  
<del\_time> is the delay time in seconds. It has the type DECIMAL.

### ***The DISTANCE Statement***

**DISTANCE,<variable\_1>,<variable\_2>**  
<variable\_1> is assigned the same value as <variable\_2> and will get the same type as the first one. The maximum length of <variable\_1> and <variable\_2> is 32 characters.  
<variable\_2> can be an expression, an integer constant or a decimal constant.  
When the DISTANCE statement is used <variable\_1> will be scaled with the current scale. This is not the case when the ASSIGN statement is used.

### ***The DRAWING\_NAME Statement***

**DRAWING\_NAME, <name\_drawing>;**  
This statement assigns the name of current drawing to <name\_drawing> which is a string parameter. If there is no current drawing, the statement prompts the user to supply the name as text either in input window or on terminal.

### ***The ELSE Statement***

**ELSE;**  
This statement must be used together with the IF statement.

### ***The ENDIF Statement***

**ENDIF;**  
The ENDIF statement terminates the IF statement.

### ***The ENDLOOP Statement***

**ENDLOOP;**  
The ENDLOOP statement terminates the LOOP statement.

### ***The ENDMACRO Statement***

**ENDMACRO;**  
The ENDMACRO statement terminates the MACRO statement.

## The **ENDWHILE** Statement

**ENDWHILE;**

The ENDWHILE statement terminates the WHILE statement.

## The **EXTRACTION** Statement (**OPTIONAL**)

**EXTRACTION,<dex\_name>,<dex\_str>;**

<dex\_name> is the name of the data extraction variable and has the type EXTRACT. The maximum length of <dex\_name> is 32 characters.

<dex\_str> is the data extraction string and has the type STRING. For further information, see documentation in *Tribon User's Guide Data Extraction*.

## The **GENERALCYLINDER** Statement

**GENERALCYLINDER,<gencyl\_name>,<cnt\_name>,<thick>,<pnt\_1>,<pnt\_2>,<pnt\_3>;**

<gencyl\_name> is the name of the general cylinder and will be assigned the type GENERALCYLINDER\_3D. The maximum length of <gencyl\_name> is 32 characters.

<cnt\_name> is the name of a 2D contour (with type CONTOUR\_2D) which has previously been defined.

<thick> is the thickness of the general cylinder and has the type DECIMAL.

<pnt\_1> is the origin of the general cylinder.

<pnt\_2> and <pnt\_3> defines together with <pnt\_1> the orientation in space. The u vector is given by <pnt\_2> and <pnt\_1> and the v vector by <pnt\_3> and <pnt\_1>.

<pnt\_1>,<pnt\_2> and <pnt\_3> all have the type POINT\_3D.

### Structure:

PNT(1:3)	(DECIMAL)
UVEC(1:3)	(DECIMAL)
VVEC(1:3)	(DECIMAL)
THICK	(DECIMAL)
NSEG	(INTEGER)
SEGPARTS(1:NSEG)	
	ENDPNT(1:2) (DECIMAL)
	AMPLITUDE(1:2) (DECIMAL)

### Example: )

```
GET/STRUCTURE=(PX,<gencyl_name>,'PNT',1
/STRUCTURE=(N,<gencyl_name>,'NSEG')
/STRUCTURE=(X,<gencyl_name>,'ENDPNT',N,1)
/STRUCTURE=(Y,<gencyl_name>,'AMPLITUDE',2,'Y');
```

## The **GET** Statement

**GET**

```
[/INTEGER=(<prompt>,<int>)]
[/DECIMAL=(<prompt>,<dec>)]
[/STRING=(<prompt>,<str>)]
[/DISTANCE=(<prompt>,<dist>)]
[/POINT_2D=(<prompt>,<pnt>)]
[/POINT_3D=(<prompt>,<pnt>)]
[/EXTRACT=(<variable>,<status>,<dex_name>,<arg_1>
[,<arg_2>...[,<arg_10>]...])]
```

```
[ /RANGE=(<rng_name>,<status>,<dex_name>[ ,<arg_1>
[ ,<arg_2>]...[ ,<arg_10>]..]])]
[ /STRUCTURE=(<variable>,<struct_name>,<arg_1>
[ ,<arg_2>]...[ ,<arg_10>]..]])];
[ /MODEL_NAME=(<prompt>, <status>, <name_model>, <name_component>)]];
```

**[/VIEW\_ID=(**<prompt>**, **<get\_viewid>**)];**

**<prompt>** is a text displayed at the workstation before entering any values. It has the type STRING and the maximum length is 100 characters. **<prompt>** has the same meaning for the following attributes to the GET statement.

**INTEGER=(**<prompt>**,**<int>**)**

**<int>** is the integer value which has been given as input at the workstation.

**DECIMAL=(**<prompt>**,**<dec>**)**

**<dec>** is the decimal value which has been given as input at the workstation.

**STRING=(**<prompt>**,**<str>**)**

**<str>** is the string value which has been given as input at the workstation.

**DISTANCE=(**<prompt>**,**<dist>**)**

**<dist>** is a distance which has been given as input at the workstation. By default, the **<dist>** is given as the distance between two cursor positions (or any other point mode) but it is possible to key in **<dist>** by answering REJECT at the first cursor position. **<dist>** has the type DECIMAL.

It is important to use DISTANCE and not DECIMAL when a distance is wanted, because the current scale is taken care of in DISTANCE but not in DECIMAL.

**POINT\_2D=(**<prompt>**,**<pnt>**)**

**<pnt>** is a point which has been given as input at the workstation. The default point mode is cursor position but it is possible to use any of the other available modes. **<pnt>** has the type POINT\_2D.

**POINT\_3D=(**<prompt>**,**<pnt>**)**

**<pnt>** is a point which has been given as input at the workstation. The default point mode is cursor position but it is possible to use any of the other available modes. **<pnt>** has the type POINT\_3D.

**EXTRACT=(**<variable>**,**<status>**,**<dex\_name>**,**<arg\_1>**[**<arg\_2>**]... [**<arg\_10>**]..)])**

**Optional function.**

**<variable>** is the name of the variable to be assigned. It can have the type INTEGER, DECIMAL or STRING. The maximum length of **<variable>** is 32 characters.

**<status>** is a variable giving the status of **<variable>**. It can have the following INTEGER values:

0**<variable>** not defined

1**<variable>** defined

Before using **<variable>**, there must always be a test on **<status>** to ensure that **<variable>** is defined. Otherwise the macro will be aborted.

**<dex\_name>** is the name of the data extraction variable and has the type EXTRACT. The maximum length of **<dex\_name>** is 32 characters.

**<arg\_1>**[**<arg\_2>**]... [**<arg\_10>**]..)]) are the arguments corresponding to the data extraction keywords, from the top level down to the bottom level. These arguments can have the type INTEGER, DECIMAL or STRING.

**RANGE=(**<rng\_name>**,**<status>**,**<dex\_name>**[**<arg\_1>**[**<arg\_2>**]... [**<arg\_10>**]..)])**

**Optional function.**

**<rng\_name>** is the name of the range which was the result of the extraction. It has the type RANGE. The maximum length of **<rng\_name>** is 32 characters. The range can for instance contain the resulting object names.

**<status>** is a variable giving the status of **<rng\_name>**. It can have the following INTEGER values:

0<rng\_name> not defined

1<rng\_name> defined

Before using <rng\_name>, there shall always be a test on <status> to ensure that <rng\_name> is defined. Otherwise the macro will be aborted.

<dex\_name> is the name of the data extraction variable and has the type EXTRACT. The maximum length of <dex\_name> is 32 characters.

[,<arg\_1>[,<arg\_2>]...[,<arg\_10>]...]] are the arguments corresponding to the data extraction keywords, from the top level down to the bottom level. These arguments can have the type INTEGER, DECIMAL or STRING.

**STRUCTURE=(<variable>,<struct\_name>,<arg\_1>[,<arg\_2>]...[,<arg\_10>]..])**

<variable> is the name of the variable to be assigned. It can have the type INTEGER, DECIMAL or STRING. The maximum length of <variable> is 32 characters.

<struct\_name> is the name of the structure from which the data shall be taken. It has the type STRING. The following structure types are available:

POINT\_2D

LINE\_2D

ARC\_2D

CONTOUR\_2D

CIRCLE\_2D

VECTOR\_2D

SPLINE\_2D

TEXTFILE\_2D

TEXT\_2D

SYMBOL\_2D

NOTE\_2D

HATCH\_2D

POINT\_3D

VECTOR\_3D

CONNECTIONPOINT\_3D

CONE\_3D

CYLINDER\_3D

GENERALCYLINDER\_3D

PARALLELEPIPED\_3D

POLYGON\_3D

SPHERESEG\_3D

TOROID\_3D

ROTATIONAL\_3D

[<arg\_1>[,<arg\_2>]...[,<arg\_10>]...]] are the arguments in the structure. It is thus possible for instance to get the end coordinates for the n:th segment in a certain contour or the total number of lines in the file used by the TEXTFILE statement.

The arguments are described at the respective structure statement.

**MODEL\_NAME=(<prompt>,<status>,<name\_model>,<name\_component>)**

<status> is a variable giving the status of <variable>. It can have the following INTEGER values:

0<name\_model> and <name\_component> not defined

1<name\_model> defined, <name\_component> not defined

2<name\_model> and <name\_component> defined

Before using either of <name\_model> and <name\_component>, there must always be a test on <status> to ensure that the variable about to be used is defined. Otherwise the macro will be aborted. Moreover, the variable must be declared by a DECLARE statement.

<name\_model> is a string assigned to by the statement. It is only valid if <status> = 1. The parameter has the name of the model indicated by user when the statement was executed.

<name\_component> is a string assigned to by the statement. It is only valid if <status> = 1. The parameter has the name of the model object component indicated by user when the statement was executed.

**VIEW\_ID=(<prompt>,<get\_viewid>)**

<get\_viewid> is the ID of a view which has been given as input at the workstation.

<get\_viewid> has the type INTEGER.

## The HATCH Statement

**HATCH,<hatch\_name>,<hatch\_cnt>,<hatch\_type>**

**[/HATCHANGLE=<hatch\_angle>]**

**[/HATCHDISTANCE=<hatch\_dist>]**

**[/USERDEFINED=(<hatch\_page>,<hatch\_number>)]**

**[/ISLAND=<island\_cnt>];**

<hatch\_name> is the name of the hatch pattern and has the type HATCH\_2D. The maximum length of <hatch\_name> is 32 characters.

<hatch\_type> is the type of hatch pattern. It has the type INTEGER. The following types are available:

1 normal, positive angle

2 normal, negative angle

3 cross-hatching

4 user defined

**HATCHANGLE=<hatch\_angle>**

<hatch\_angle> is the angle of the hatch pattern lines. It has the type DECIMAL. The default angle is 60 degrees.

**HATCHDISTANCE=<hatch\_dist>**

<hatch\_dist> is the distance between the hatch pattern lines. It has the type DECIMAL. The default distance is 5 mm.

**USERDEFINED=<hatch\_page>,<hatch\_number>**

<hatch\_page> is the page number (1-999) in the standard book for user defined hatch patterns. It has the type INTEGER.

<hatch\_number> is the standard number (1-8) within the page <hatch\_page> in the standard book for user defined hatch patterns. It has the type INTEGER.

**/ISLAND=<island\_cnt>**

<island\_cnt> is the island contour where the hatch pattern shall be removed. It is of the type CONTOUR\_2D or TEXT\_2D.

### Structure:

TYPE	(INTEGER)
ANGLE	(DECIMAL)
DISTANCE	(DECIMAL)
PAGE	(INTEGER)
NUMBER	(INTEGER)
NSEG	(INTEGER)
ISLANDS	(INTEGER)
SEGPARTS ( 1 : NSEG )	
	ENDPNT ( 1 : 2 ) (DECIMAL)
	AMPLITUDE ( 1 : 2 ) (DECIMAL)

### Example:



```

GET/STRUCTURE=(T,<hatch_name>,'TYPE'
/STRUCTURE=(A,<hatch_name>,'ANGLE')
/STRUCTURE=(D,<hatch_name>,'DISTANCE')
/STRUCTURE=(P,<hatch_name>,'PAGE')
/STRUCTURE=(M,<hatch_name>,'NUMBER')
/STRUCTURE=(N,<hatch_name>,'NSEG')
/STRUCTURE=(I,<hatch_name>,'ISLANDS')
/STRUCTURE=(X,<hatch_name>,'ENDPNT',N,1)
/STRUCTURE=(Y,<hatch_name>,'AMPLITUDE',2,'Y');

```

## The IF Statement

**IF,<cond>;**

<cond> is the condition to be tested. It has the type

BOOLEAN.

## The LAYER Statement

**LAYER,<lay\_no>;**

<lay\_no> is the layer number (> 0) and must be given within '' if it is addressed with an alias name. If <lay\_no> is a true number or a variable, it shall not be given within ''. If <lay\_no> refers to a layer class, it must be preceded by # and directly followed by the class number or the class name. In this case, <lay\_no> shall be given within ''.

The maximum length of <lay\_no> as an alias is 26 characters and as a variable 32 characters.

## The LINE Statement

**LINE,<line\_name>,<stp\_pnt>**

**[/LINEEND=<end\_pnt>]**

**[/LINEANGLE=(<len>,<ang>)]**

**[/LINEOFFS=<offs>];**

<line\_name> is the name of the line and has the type LINE\_2D. The maximum length of <line\_name> is 32 characters.

<stp\_pnt> is the starting point of the line with the type POINT\_2D.

**LINEEND=<end\_pnt>**

<end\_pnt> is the ending point of the line with the type POINT\_2D.

**LINEANGLE=(<len>,<ang>)**

<len> is the length of the line and <ang> is the angle of the line, both with type DECIMAL.

**LINEOFFS=<offs>**

<offs> is the offset from the current point. It has the type STRING with at most 72 characters.

The following formats are valid:

du, dv (e.g. 100,50) or

length, angle (e.g. 100,45D) or

length, verbal direction (e.g. 100,N).

Verbal directions can be North, South, West, East, Right, Left, Up or Down. Only the first letter is relevant. If the string ends with a 'U' or a 'V', the u axis or the v axis will be locked. Angles must be followed by D (degrees) or R (radians).

The strings must only contain digits, a '-' or a '!'.'-' is only valid in the first position. Between the strings, it is allowed to have a couple of '' and ',' but at least one, either a '' or a ','. If one of the letters mentioned above is at a valid position, the rest of the string is ignored. Spaces in the beginning or the end of the string are removed.

**Structure:**

STARTPNT(1:2) (DECIMAL)

ENDPNT(1:2) (DECIMAL)

#### Example:

```
GET/STRUCTURE=(X,<line_name>,'STARTPNT',1)
/STRUCTURE=(Y,<line_name>,'ENDPNT','Y')
```

### The LINETYPE Statement

**LINETYPE,<lin\_type>**

**[/LINEWIDTH=<lin\_wid>];**

<lin\_type> is the modal line type to be used. It has the type INTEGER. Default line type is 1 (solid). The types 1 - 5 are available.

**LINEWIDTH=<lin\_wid>**

<lin\_wid> is the modal line width to be used. It has the type INTEGER. Default line width is 1 (thin). The types 1 - 3 are available.

### The LOOP Statement

**LOOP,<loop\_var>,<range>;**

<loop\_var> is the loop variable and the type is dependent on how <range> is given.

<range> can be given in three ways:

<start\_value> : <end\_value>

or

<start\_value> : <end\_value> :: <step\_value>

or

<range>

where <range> has the type RANGE.

In the to first cases <loop\_var> can have the type INTEGER or DECIMAL. If <range> has the type RANGE then <loop\_var> can also have type STRING.

When <step\_value> is omitted, it is by default put to 1.

<start\_value>, <end\_value> and <step\_value> have the type DECIMAL.

### The MACRO Statement

**MACRO,<macro\_name>**

**[,<arg\_1>,<arg\_2>...,<arg\_25>]....];**

<macro\_name> is the name of the macro. The maximum length of <macro\_name> is 32 characters. Only upper case letters, digits and \_ are allowed characters. <arg\_1>, <arg\_2>, ..., <arg\_25> are the arguments to the macro. They cannot be expressions but must be variables. All types given in the ASSIGN statement are allowed.

### The NAME Statement

**NAME,<dwg\_name>**

**[/DWG]**

**[/FORM=<form\_name>]**

**[/PICT]**

**[/SCALE=<dwg\_scale>]**

**[/VOLUME=(<xmax>,<ymin>)];**

<dwg\_name> is the name of the drawing/subpicture/ volume to be created and stored on a data bank. If the NAME statement is omitted, <macro\_name> will be used. <dwg\_name> has the type STRING and the maximum length is 32 characters.

If the drawing/subpicture/volume given by <dwg\_name> already exists on the data bank, the geometry created by the macro will be added to the existing object.

## DWG

This attribute is used to specify that a drawing is to be created.

## FORM=<form\_name>

<form\_name> is the name of the form to be used, if any. <form\_name> has the type STRING and the maximum length is 32 characters.

## PICT

This attribute is used to specify that a subpicture is to be created.

## SCALE=<dwg\_scale>

<dwg\_scale> is the drawing/subpicture scale to be used. The default scale is 1:50.

<dwg\_scale> has the type DECIMAL.

## VOLUME=(<xmax>,<yman>)

<xmax> and <yman> are the x and y extensions for the volume to be created. They have the type DECIMAL.

## The NOTE Statement

**NOTE,<note\_name>,<start\_pnt>,<ref\_cnt>,<note\_text>**

**[/NOTESYMBOL=<note\_symb>]**

**[/REFSYMBOL=<ref\_symb>];**

<note\_name> is the name of the note and has the type NOTE\_2D. The maximum length of <note\_name> is 32 characters.

<start\_pnt> is the starting point of the reference lines of the note. It has the type POINT\_2D.

<ref\_cnt> is the reference lines of the note. It has the type CONTOUR\_2D. This contour must have been defined previously.

<note\_text> is the text in the note (with type TEXT\_2D) which has previously been defined.

## NOTESYMBOL=<note\_symb>

<note\_symb> is the number of the note symbol. It has the type INTEGER. The default note symbol is number 31.

## REFSYMBOL=<ref\_symb>

<ref\_symb> is the number of the reference symbol. It has the type INTEGER. The default reference symbol is number 21.

## Structure:

TEXT	(STRING)
NOTESYMB	(INTEGER)
REFSYMB	(INTEGER)
PNT(1:2)	(DECIMAL)
NSEG	(INTEGER)
SEGPARTS(1:NSEG)	
	ENDPNT(1:2) (DECIMAL)
	AMPLITUDE(1:2) (DECIMAL)

### Example:

```
GET/STRUCTURE=(NS,<note_name>,'NOTESYMB')
/STRUCTURE=(RS,<note_name>,'REFSYMB')
/STRUCTURE=(SX,<note_name>,'PNT',1)
/STRUCTURE=(N,<note_name>,'NSEG')
/STRUCTURE=(X,<note_name>,'ENDPNT',N,1)
/STRUCTURE=(Y,<note_name>,'AMPLITUDE',2,'Y');
```

## The PARALLELEPIPED Statement

**PARALLELEPIPED,<para\_name>**

**[/COORDPARA=(<corn\_1>,<corn\_2>,<corn\_3>)]**

**[/CENLINPARA=(<cl\_pnt\_1>,<cl\_pnt\_2>,<corn>)];**

<para\_name> is the name of the parallelepiped and has the type PARALLELEPIPED\_3D. The maximum length of <para\_name> is 32 characters.

**COORDPARA=(<corn\_1>,<corn\_2>,<corn\_3>)**

<corn\_1> is the origin of the parallelepiped and the lower right corner when looking in the direction of the centre line. <corn\_2> is the upper right corner at the end surface when looking in the same direction. <corn\_3> is then, in the same way, the upper left corner at the end surface.

<corn\_1>, <corn\_2> and <corn\_3> all have the type POINT\_3D.

**COORDPARA=(<cl\_pnt\_1>,<cl\_pnt\_2>,<corn>)**

<cl\_pnt\_1> is the starting point of the centre line and <cl\_pnt\_2> is the ending point of the centre line.

<corn> is the upper left corner at the end surface when looking in the direction of the centre line.

<cl\_pnt\_1>, <cl\_pnt\_2> and <corn> all have the type POINT\_3D.

#### Structure:

PNT(1:3) (DECIMAL)  
UVEC(1:3) (DECIMAL)  
VVEC(1:3) (DECIMAL)  
LENGTH (DECIMAL)

#### Example:

```
GET/STRUCTURE=(X,<para_name>,'PNT',1)  
/STRUCTURE=(Y,<para_name>,'UVEC',2)  
/STRUCTURE=(Z,<para_name>,'VVEC','Z')  
/STRUCTURE=(L,<para_name>,'LENGTH');
```

### The POINT\_2D Statement

**POINT\_2D,<pnt\_name>,<x>,<y>;**

<pnt\_name> is the name of the point and has the type POINT\_2D. The maximum length of <pnt\_name> is 32 characters.

<x> and <y> are the x and y coordinates of the point, both with type DECIMAL.

If only one of the coordinates of a previously defined point are to be changed, the other one is simply omitted.

#### Structure:

PNT(1:2) (DECIMAL)

#### Example:

```
GET/STRUCTURE=(X,<pnt_name>,'PNT',1)  
/STRUCTURE=(Y,<pnt_name>,'PNT','Y');
```

### The POINT\_3D Statement

**POINT\_3D,<pnt\_name>,<x>,<y>,<z>;**

<pnt\_name> is the name of the point and has the type POINT\_3D. The maximum length of <pnt\_name> is 32 characters.

<x>, <y> and <z> are the x, y and z coordinates of the point, all with type DECIMAL.

If only one of the coordinates of a previously defined point shall be changed, the others are simply omitted.

**Structure:**

PNT(1:3) (DECIMAL)

**Example:**

```
GET/STRUCTURE=(X,<pnt_name>,'PNT','X')
/STRUCTURE=(Y,<pnt_name>,'PNT',2)
/STRUCTURE=(Z,<pnt_name>,'PNT','Z');
```

### The POLYGON Statement

**POLYGON,<pol\_name>,<stp\_pnt>**

**/LINEPOLYGON=<end\_pnt>....;**

<pol\_name> is the name of the polygon and has the type POLYGON\_3D. The maximum length of <pol\_name> is 32 characters.

<stp\_pnt> is the starting point of the polygon with the type POINT\_3D.

**LINEPOLYGON=<end\_pnt>....**

<end\_pnt> is the ending point of the polygon segment with the type POINT\_3D.

**Structure:**

NSEG (INTEGER)  
SEGPARTS(1:NSEG)  
ENDPNT(1:3) (DECIMAL)  
AMPLITUDE(1:3) (DECIMAL)

**Example:**

```
GET/STRUCTURE=(N,<pol_name>,'NSEG')
/STRUCTURE=(X,<pol_name>,'ENDPNT',N,1)
/STRUCTURE=(Y,<pol_name>,'AMPLITUDE',2,'Y');
```

### The PRESENT Statement

**PRESENT,<var\_name>;**

<var\_name> is the name of the variable which is to be presented. It can have any of the following types:

POINT\_2D

LINE\_2D

ARC\_2D

CIRCLE\_2D

CONTOUR\_2D

SPLINE\_2D

SYMBOL\_2D

TEXT\_2D

TEXTFILE\_2D

NOTE\_2D

TABLE

CONNECTIONPOINT\_3D

CONE\_3D

CYLINDER\_3D

PARALLELEPIPED\_3D  
POLYGON\_3D  
ROTATIONAL\_3D  
TOROID\_3D  
SPHERESEG\_3D  
GENERALCYLINDER\_3D

The maximum length of <var\_name> is 32 characters.

### The PUT Statement

**PUT,<var\_name>;**

<var\_name> is the name of the variable which contents is to be written. It can have any type. The maximum length of <var\_name> is 32 characters.

### The RANGE Statement

**RANGE,<rng\_name>**

**[/FILENAME=<file\_name>];**

<rng\_name> is the name of the range and has the type RANGE. The maximum length of <rng\_name> is 32 characters.

**FILENAME=<file\_name>**

<file\_name> is the file where the range is stored. It has the type STRING. The file contains a number of integer numbers OR decimal numbers OR strings, one on each line.

### The ROTATIONAL Statement

**ROTATIONAL,<rot\_name>,<cnt\_name>,<pnt>,<vec>;**

<rot\_name> is the name of the rotational primitive and will be assigned the type ROTATIONAL\_3D. The maximum length of <rot\_name> is 32 characters.

<cnt\_name> is the name of a 2D contour (with type CONTOUR\_2D) which has previously been defined.

<pnt> is the origin of the 2D contour in space. It has the type POINT\_3D.

<vec> is the vector around which <cnt\_name> is rotated. It has the type VECTOR\_3D.

#### Structure:

PNT(1:3)	(DECIMAL)
VEC(1:3)	(DECIMAL)
NSEG	(INTEGER)
SEGPARTS(1:NSEG)	
	ENDPNT(1:2) (DECIMAL)
	AMPLITUDE(1:2) (DECIMAL)

#### Example:

```
GET/STRUCTURE=(X,<rot_name>,'PNT',1)
/STRUCTURE=(N,<rot_name>,'NSEG')
/STRUCTURE=(Y,<rot_name>,'AMPLITUDE',N,'Y');
```

### The SPHERESEG Statement

**SPHERESEG,<seg\_name>,<rad>**

**/COORDSEG=(<cl\_pnt\_1>,cl\_pnt\_2>;**

<seg\_name> is the name of the spherical segment and has the type SPHERESEG\_3D. The maximum length of <seg\_name> is 32 characters.

<rad> is the radius of the spherical segment. It has the type DECIMAL.

**COORDSEG=(<cl\_pnt\_1>,<cl\_pnt\_2>)**

<cl\_pnt\_1> is the starting point and <cl\_pnt\_2> is the ending point of the centre line, both with the type POINT\_3D.

**Structure:**

PNT(1:3)       (DECIMAL)  
UVEC(1:3)       (DECIMAL)  
RADIUS         (DECIMAL)

**Example:**

```
GET/STRUCTURE=(X,<seg_name>,'PNT',1)  
/STRUCTURE=(Z,<seg_name>,'UVEC',3)  
/STRUCTURE=(R,<seg_name>,'RADIUS');
```

### The SPLINE Statement

**SPLINE,<spl\_name>**

**[/SPLPNT=<spl\_pnt>]**

**[/SPLTAN=(<spl\_pnt>,<tang\_ang>)];**

<spl\_name> is the name of the spline and has the type SPLINE\_2D. The maximum length of <spl\_name> is 32 characters.

It is possible to define the spline with or without tangent conditions in the points in the following way:

- 1)no tangent condition
- 2)tangent condition in the start point
- 3)tangent condition in the end point
- 4)tangent conditions in both start and end point
- 5)tangent conditions in all points

**SPLPNT=<spl\_pnt>**

<spl\_pnt> is the spline point with type POINT\_2D. It is used when no tangent condition is wanted.

**SPLTAN=(<spl\_pnt>,<tang\_ang>)**

<spl\_pnt> is the spline point with type POINT\_2D.

<tang\_ang> is the tangent angle to be used in 2) - 5) above. It has the type DECIMAL.

**Structure:**

NSEG                   (INTEGER)  
SEGPARTS(1:NSEG)  
PNT(1:2)       (DECIMAL)  
VEC(1:2)       (DECIMAL)

**Example: )**

```
GET/STRUCTURE=(N,<spl_name>,'NSEG'  
/STRUCTURE=(X,<spl_name>,'PNT',N,1)  
/STRUCTURE=(Y,<spl_name>,'VEC',N,2);
```

### The SPLIT Statement

**SPLIT,<model\_name>,<delimiter>,<project>,<module>,<subsys\_1>,<subsys\_2>,<subsys\_3>;**

<model\_name> is the name of the model. It has the type STRING. The maximum length of <col> is 26 characters.

<delimiter> is the character which is used to separate the different parts of <model\_name>. It has the type STRING.

<project> is the project of <model\_name> and has the type STRING.

<module> is the module of <model\_name> and has the type STRING.

<subsys1\_1> is the first subsystem of <model\_name> and has the type STRING.

<subsys1\_2> is the second subsystem of <model\_name> and has the type STRING.

<subsys1\_3> is the third subsystem of <model\_name> and has the type STRING.

All three subsystems need not be present in the model name. How many subsystems that exist depends on the application.

## The SYMBOL Statement

**SYMBOL,<symb\_name>,<symb\_no>,<symb\_pnt>,<symb\_ang>**

**[/MIRRORX]**

**[/MIRRORY]**

**[/SYMBFONT=<symb\_fnt>]**

**[/SYMBHEIGHT=<symb\_hgt>]**

**[/SYMBWIDTH=<symb\_wid>]**

**[/AUTOOFF];**

<symb\_name> is the name of the symbol and has the type SYMBOL\_2D. The maximum length of <symb\_name> is 32 characters.

<symb\_no> is the number of the symbol. It has the type INTEGER.

<symb\_pnt> is the origin of the symbol and has the type POINT\_2D.

<symb\_ang> is the symbol angle which is of type DECIMAL.

### **MIRRORX**

With this attribute, it is possible to reflect the symbol in the x axis. It cannot be used together with MIRRORY.

### **MIRRORY**

With this attribute it is possible to reflect the symbol in the y axis. It cannot be used together with MIRRORX.

### **SYMBFONT=<symb\_fnt>**

<symb\_fnt> is the symbol font and has the type INTEGER. The default font is the system font number 21.

### **SYMBHEIGHT=<symb\_hgt>**

<symb\_hgt> is the symbol height and has the type DECIMAL. The symbol is given a default height value and the symbol height can be changed with this attribute.

### **SYMBWIDTH=<symb\_wid>**

<symb\_wid> is the symbol width and has the type DECIMAL. The symbol is given a default width value but the symbol width can be changed with this attribute.

### **AUTOOFF**

With this attribute it is possible to switch off the default automatic positioning of the symbol.

## Structure:

SYMBNO	(INTEGER)
PNT ( 1 : 2 )	(DECIMAL)
ANGLE	(DECIMAL)
FONTNO	(INTEGER)
WIDTH	(DECIMAL)
HEIGHT	(DECIMAL)
MIRR	(INTEGER)
AUTO	(INTEGER)



**Example:**

```

GET/STRUCTURE=(S,<symb_name>,'SYMBNO')
/STRUCTURE=(X,<symb_name>,'PNT',1)
/STRUCTURE=(A,<symb_name>,'ANGLE')
/STRUCTURE=(F,<symb_name>,'FONTNO')
/STRUCTURE=(W,<symb_name>,'WIDTH')
/STRUCTURE=(H,<symb_name>,'HEIGHT')
/STRUCTURE=(M,<symb_name>,'MIRR')
/STRUCTURE=(A,<symb_name>,'AUTO');

```

**The TABLE Statement**

```

TABLE,<tab_name>
[/ARGUMENTS=([<arg_1>,<arg_2>....,<arg_15>]....)]
[/FORMAT=([<width_1>,<ndec_1>,<width_2>,<ndec_2>....
[,<width_15>,<ndec_15>]....])]
[/FILENAME=<file_name>]
[/QUOTES]
[/NOAPPEND];

```

<tab\_name> is the name of the table and has the type TABLE. The maximum length of <tab\_name> is 32 characters.

**ARGUMENTS=([<arg\_1>,<arg\_2>.... [<arg\_15>]....])**

<arg\_1>,<arg\_2>,...,<arg\_15> are the arguments to be put in the table. They can have the type INTEGER, DECIMAL or STRING.

**FORMAT=([<width\_1>,<ndec\_1>,<width\_2>,<ndec\_2>.... [<width\_15>,<ndec\_15>]....])**

This attribute describes the format of the table. <width\_1>,<width\_2>,...,<width\_15> is the width of the column for the i:th argument. It has the type integer. The default width is 8 which will be the result if FORMAT is not given or if the width for a variable is omitted (i.e. given as ,). If the width is negative the result will be positioned left-justified, otherwise right-justified.

<ndec\_1>,<ndec\_2>,...,<ndec\_15> is the number of decimals for the i:th argument. It has the type integer. The default number of decimals is 3 which will be the result if FORMAT is not given. It is of course only necessary to give the number of decimals for arguments of type DECIMAL. In other cases it can be omitted (i.e. given as ,) or given as 0.

The attributes ARGUMENTS and FORMAT must be given together and with the same number of parameters.

**FILENAME=<file\_name>**

<file\_name> is the file where the table will be presented when the PRESENT statement is used. It has the type STRING.

**QUOTES**

When this attribute is given in the TABLE statement all terms will be surrounded by '. The resulting file can then directly be used as input to the Tribon Report Generator.

**NOAPPEND**

When this attribute is given in the TABLE statement the old contents in <tab\_name> will be deleted and the <tab\_name> can be used for another table. The attribute must also be given the first time <tab\_name> is used.

**The TEXT Statement**

```

TEXT,<txt_name>,<txt_pnt>
/TEXTLINE=<txt_str>]
/TEXTDECNO=<txt_dec>]
/TEXTINTNO=<txt_int>]
/TEXTANGLE=<txt_ang>]

```

**[/TEXTHEIGHT=<txt\_hgt>]**

**[/TEXTFONT=<txt\_fnt>];**

<txt\_name> is the name of the text line and has the type TEXT\_2D. The maximum length of <txt\_name> is 32 characters.

<txt\_pnt> is the text origin which is the lower left corner of the text. It has the type POINT\_2D.

**TEXTDECNO=<txt\_dec>**

<txt\_dec> is a number which shall be presented as a text. It has the type DECIMAL.

**TEXTINTNO=<txt\_int>**

<txt\_int> is a number which will be presented as a text. It has the type INTEGER.

**TEXTLINE=<txt\_str>**

<txt\_str> is a string which will be presented as a text. It has the type STRING.

**TEXTANGLE=<txt\_ang>**

<txt\_ang> is the text angle and has the type DECIMAL. The default angle is 0.0 but the text angle can be changed with this attribute.

**TEXTHEIGHT=<txt\_hgt>**

<txt\_hgt> is the text height and has the type DECIMAL. The text is given a default height value but the text height can be changed with this attribute.

**TEXTFONT=<txt\_fnt>**

<txt\_fnt> is the text font and has the type INTEGER. Available font numbers are 0-99. The default font number is 0.

#### Structure:

TEXT	(STRING)
PNT(1:2)	(DECIMAL)
ANGLE	(DECIMAL)
HEIGHT	(DECIMAL)

#### Example:

```
GET/STRUCTURE=(T,<txt_name>,'TEXT'  
/STRUCTURE=(X,<txt_name>,'PNT',1)  
/STRUCTURE=(A,<txt_name>,'ANGLE')  
/STRUCTURE=(H,<txt_name>,'HEIGHT');
```

### The TEXTFILE Statement

**TEXTFILE,<txf\_name>,<file\_name>**

**[/INTERLINESPACE=<intlin\_space>]**

**[/POSINES=(<start\_pnt>,<first\_line>,<last\_line>)]**

**[/TEXTANGLE=<text\_ang>]**

**[/TEXTHEIGHT=<text\_hgt>]**

**[/TEXTFONT=<text\_fnt>];**

<txf\_name> is the name of the text file variable and has the type TEXTFILE\_2D. The maximum length of <txf\_name> is 32 characters.

<file\_name> is the name of the file from where the text is fetched. It has the type STRING.

**INTERLINESPACE=<intlin\_space>**

<intlin\_space> is the factor for the interline space. The interline space is defined as the factor \* the text height. It has the type DECIMAL. The default factor is 1.5.

**POSINES=(<start\_pnt>,<first\_line>,<last\_line>)**

<start\_pnt> is the origin of the first text line and the lower left corner of that line. It has the type POINT\_2D.

<first\_line> is the number of the first line to be presented. It has the type INTEGER.

<last\_line> is the number of the last line to be presented. It has the type INTEGER.

**TEXTANGLE=<text\_ang>**

<text\_ang> is the text angle and has the type DECIMAL. The default angle is 0.0 but the text angle can be changed with this attribute.

**TEXTHEIGHT=<text\_hgt>**

<text\_hgt> is the next height and has the type DECIMAL. The text is given a default height value but the text height can be changed with this attribute.

**TEXTFONT=<text\_fnt>**

<text\_fnt> is the text font and has the type INTEGER. Available font numbers are 0-99. The default font number is 0.

It is possible to give both absolute and relative values of the lines to be presented. This is illustrated by the following examples:

/POSLINES=(P1,1,15) lines 1 to 15

/POSLINES=(P1,10,) next 10 lines (16 to 25)

/POSLINES=(P1,20,) next 20 lines (26 to 45)

/POSLINES=(P1,,10) previous 10 lines (16 to 25)

/POSLINES=(P1,,) the whole file

**Structure:**

FILENAME	(STRING)
ANGLE	(DECIMAL)
HEIGHT	(DECIMAL)
ILSP	(DECIMAL)
NDATA	(INTEGER)
DATA ( 1 : NDATA )	
PNT ( 1 : 2 )	(DECIMAL)
FIRST	(INTEGER)
LAST	(INTEGER)

**Example:**

```
GET/STRUCTURE=(F,<txf_name>,'FILENAME'  
/STRUCTURE=(A,<txf_name>,'ANGLE'  
/STRUCTURE=(H,<txf_name>,'HEIGHT'  
/STRUCTURE=(I,<txf_name>,'ILSP'  
/STRUCTURE=(N,<txf_name>,'NDATA'  
/STRUCTURE=(X,<txf_name>,'PNT',N,1)  
/STRUCTURE=(S,<txf_name>,'FIRST',N)  
/STRUCTURE=(E,<txf_name>,'LAST',N);
```

## The TOROID Statement

**TOROID,<tor\_name>,<rad>**

**/COORDTOR=<cl\_pnt\_1>,<cl\_pnt\_2>,<cl\_pnt\_3>;**

<tor\_name> is the name of the toroid and has the type TOROID\_3D. The maximum length of <tor\_name> is 32 characters.

<rad> is the radius of the toroid. It has the type DECIMAL.

**/COORDTOR=<cl\_pnt\_1>,<cl\_pnt\_2>,<cl\_pnt\_3>;**

<cl\_pnt\_1> is the starting point, <cl\_pnt\_2> is the mid point and <cl\_pnt\_3> is the ending point of the toroid centre line. All have the type POINT\_3D.

**Structure:**

```
STARTPNT(1:3)    (DECIMAL)
ENDPNT(1:3)      (DECIMAL)
AMPLITUDE(1:3)   (DECIMAL)
```

**Example:**

```
GET/STRUCTURE=(X,<tor_name>,'STARTPNT',1)
/STRUCTURE=(Y,<tor_name>,'ENDPNT',2)
/STRUCTURE=(Z,<tor_name>,'AMPLITUDE',3);
```

### The VECTOR\_2D Statement

**VECTOR\_2D,<vect\_name>,<x>,<y>;**

<vect\_name> is the name of the vector and has the type VECTOR\_2D. The maximum length of <vect\_name> is 32 characters.

<x> and <y> are the x and y coordinates of the vector, both with type DECIMAL.

If only one of the coordinates of a previously defined point shall be changed, the other one is simply omitted.

**Structure:**

```
VEC(1:2) (DECIMAL)
```

**Example: )**

```
GET/STRUCTURE=(X,<vect_name>,'VEC',1
/STRUCTURE=(Y,<vect_name>,'VEC','Y');
```

### The VECTOR\_3D Statement

**VECTOR\_3D,<vect\_name>,<x>,<y>,<z>;**

<vect\_name> is the name of the vector and has the type VECTOR\_3D. The maximum length of <vect\_name> is 32 characters.

<x>, <y> and <z> are the x, y and z coordinates of the vector, all with type DECIMAL.

If only one of the coordinates of a previously defined point shall be changed, the others are simply omitted.

**Structure:**

```
VEC(1:3) (DECIMAL)
```

**Example: )**

```
GET/STRUCTURE=(X,<vect_name>,'VEC',1
/STRUCTURE=(Y,<vect_name>,'VEC','Y')
/STRUCTURE=(Z,<vect_name>,'VEC','Z');
```

### The VMSJOB Statement

**VMSJOB,<jobfilename>;**

The parameter <jobfilename> is the name of the file to be executed. It has the type STRING. The maximum length of <jobfilename> is 100 characters.

### The WHILE Statement

**WHILE,<cond>;**

<cond> is the condition to be tested. It has the type BOOLEAN.

## 17.4.2 Example of Macros Creating 2D Geometry

## Example 1

This example is a macro creating some simple geometry.

```
!  
! All macros begin with the MACRO-stmt  
!  
MACRO,RECTANGLE;  
  GET/POINT_2D=('Give first corner',P1)  
    /POINT_2D=('Give second corner',P3);  
  ASSIGN,X1,P1/XCOORD;  
  ASSIGN,Y1,P1/YCOORD;    ! The x- and y-coordinates  
  ASSIGN,X2,P3/XCOORD;    ! of the 2 points are needed  
  ASSIGN,Y2,P3/YCOORD;  
  POINT_2D,P2,X1,Y2;      ! Create the other  
  POINT_2D,P4,X2,Y1;      ! two corners  
  CONTOUR,CNT,P1  
    /LINEEND = P2  
    /LINEEND = P3          ! Create a contour  
    /LINEEND = P4  
    /LINEEND = P1;  
  PRESENT,CNT;            ! Display the contour  
  !  
  ! All macros end with the ENDMACRO-stmt  
  !  
ENDMACRO;
```

In the macro RECTANGLE, the contour was created using the four 2D points. In some cases, it might be more convenient to define a contour with lines instead. The macro above will then look like this:

```
MACRO,RECTANGLE;  
  GET/POINT_2D=('Give first corner',P1)  
    /POINT_2D=('Give second corner',P3);  
  ASSIGN,X1,P1/XCOORD;  
  ASSIGN,Y1,P1/YCOORD;  
  ASSIGN,X2,P3/XCOORD;  
  ASSIGN,Y2,P3/YCOORD;  
  POINT_2D,P2,X1,Y2;  
  POINT_2D,P4,X2,Y1;  
  LINE,L1,P1/LINEEND=P2;  
  LINE,L2,P2/LINEEND=P3;  
  LINE,L3,P3/LINEEND=P4;  
  LINE,L4,P4/LINEEND=P1;  
  CONTOUR,CNT  
    /LINE = L1  
    /LINE = L2  
    /LINE = L3  
    /LINE = L4;
```

```
PRESENT , CNT ;  
ENDMACRO ;
```

Note that the CONTOUR statement looks different now. In the first case, the start point must be given but in the second case this point is the start point of the first line.

## Example 2


The macro SCREW below has 4 parameters. One of them, the drawing name, is given in the MACRO statement. This parameter tells where the result of the macro shall be stored when executed by the stand alone program. A drawing will be created using the form A3 and the scale chosen is 1:1. The NAME statement has no effect when the macro is executed interactively at a workstation. The result is then put into the current drawing.

The macro can be found by following this link:



[macro\\_screw.txt](#)

The figure below shows the result of the execution of the macro SCREW. The current drawing contained the rectangle and the length of the screw was given by using the different point modes, in this case the node mode. The start point was given with the close point mode.



*Figure 17:1. The result of the execution of the macro SCREW.*



## 17.4.3 A Macro Generating a Volume

### Example 3


The macro PUMP is an example of a geometry macro creating a volume. It uses most of the volume primitives. Note that the general cylinder has a 2D parameter but the result is 3D. If the macro is run stand alone, the NAME statement gives the name of the volume model on the data bank. The x- and y-extensions are also given.

The macro can be found by following this link:



[macro\\_pump.txt](#)

The figure below shows the result of the execution of the macro PUMP. The input angles were 180 degrees and 45 degrees respectively. The dimension was given as the distance between two cursor positions, both for the pump length and the pump height.



*Figure 17:2. The result of the execution of the macro PUMP.*

```

MACRO, PUMP;
NAME, 'MDORINGPUMP' /VOLUME=(3000, 3000);
GET/DECIMAL =(' Angle for pressure side: ', ANG1)
/DECIMAL =(' Angle for suction side: ', ANG2)
/POINT_3D =(' Start point: ', STP)
/DISTANCE =(' Pump length: ', LEN)
/DISTANCE =(' Pump height: ', DIA);
ASSIGN, X, STP/XCOORD;
ASSIGN, Y, STP/YCOORD;
ASSIGN, Z, STP/ZCOORD;
ASSIGN, L, LEN%1250.0;
ASSIGN, R, DIA%400.0%2;
ASSIGN, L1, 100*L;
ASSIGN, L2, 405*L;
ASSIGN, L3, 50*L;
ASSIGN, L4, 15*L;
ASSIGN, L5, 270*L;
ASSIGN, L6, L4;
ASSIGN, L7, L4;
ASSIGN, L8, 305*L;
ASSIGN, L9, L4;
ASSIGN, L10, 25*L;
ASSIGN, A11, 35*L;
ASSIGN, L12, 330*L;
ASSIGN, H12, 10%330.0*L12;
ASSIGN, W12, 80%330.0*L12;
ASSIGN, L13, L12;
ASSIGN, H13, H12;
ASSIGN, W13, W12;
ASSIGN, L14, 75*L;
ASSIGN, L15, L14;
ASSIGN, L16, L4;
ASSIGN, L17, L4;
ASSIGN, L18, L4;
ASSIGN, L19, L4;
ASSIGN, L20, 125*L;
ASSIGN, L21, L4;
ASSIGN, L22, L4;
ASSIGN, L23, L3;
ASSIGN, L24, L3;
ASSIGN, L25, L4;
ASSIGN, L26, L20;
ASSIGN, L27, L4;
ASSIGN, L28, L4;
ASSIGN, L29, L14;
ASSIGN, L30, L4;
ASSIGN, R1A, 300*R;
ASSIGN, R1B, 350*R;
ASSIGN, R2, R1B;
ASSIGN, R3, R1A;
ASSIGN, R4, 400*R;
ASSIGN, R5, 375*R;
ASSIGN, R7, R1A;
ASSIGN, R8, 225*R;
ASSIGN, R9, R8;
ASSIGN, R10, 150*R;
ASSIGN, R11, R10;
ASSIGN, R14A, 75*R;
ASSIGN, R14B, 100*R;
ASSIGN, R15A, R14A;
ASSIGN, R15B, R14B;
ASSIGN, R16, 200*R;
ASSIGN, R17, R16;
ASSIGN, R18, R16;

```

```

ASSIGN, R19, R16;
ASSIGN, R20, 60*R;
ASSIGN, R21, R10;
ASSIGN, R22, R10;
ASSIGN, R23A, 10*R;
ASSIGN, R23B, 125*R;
ASSIGN, R24A, R23A;
ASSIGN, R24B, R23B;
ASSIGN, R26, R20;
ASSIGN, R25, R10;
ASSIGN, R27, R10;
ASSIGN, R28, R10;
ASSIGN, R29, R10;
ASSIGN, R30, R10;
ASSIGN, XE1, X+L1;
ASSIGN, YE1, Y;
ASSIGN, ZE1, Z;
VECTOR_3D, VEC1, 1, 0, 0;
POINT_3D, PE1, XE1, YE1, ZE1;
CONE, CON1, R1A, R1B/COORDCONE=(P, PE1);
COLOUR, ' RED' ;
PRESENT, CON1;
ASSIGN, XE2, L2+XE1;
POINT_3D, PE2, XE2, YE1, ZE1;
CYLINDER, CYL1, R2/COORDCYL=(PE1, PE2);
PRESENT, CYL1;
ASSIGN, XE3, L3+XE2;
POINT_3D, PE3, XE3, YE1, ZE1;
CYLINDER, CYL2, R3/COORDCYL=(PE2, PE3);
PRESENT, CYL2;
ASSIGN, XE4, L4+XE3;
POINT_3D, PE4, XE4, YE1, ZE1;
ASSIGN, R42, R4%2;
POINT_2D, PC1, R4, R42;
POINT_2D, PC2, R4, -R42;
POINT_2D, PC3, -R4, -R42;
POINT_2D, PC4, -R4, R42;
POINT_2D, PC5, 0, 1.5*R4;
POINT_3D, PGC1, XE3, YE1, ZE1-R42;
POINT_3D, PGC2, XE3, YE1+1, ZE1-R42;
POINT_3D, PGC3, XE3, YE1, ZE1+1-R42;
CONTOUR, CNT1, PC1
/LINEEND=PC2
/LINEEND=PC3
/LINEEND=PC4
/ARCMIDPNT=(PC5, PC1);
GENERALCYLINDER, GCL1, CNT1, L4, PGC1, PGC2, PGC3;
COLOUR, ' GREEN' ;
PRESENT, GCL1;
ASSIGN, XE5, L5+XE4;
POINT_3D, PE5, XE5, YE1, ZE1;
CYLINDER, CYL3, R5/COORDCYL=(PE4, PE5);
COLOUR, ' MAGENTA' ;
PRESENT, CYL3;
ASSIGN, XE6, L6+XE5;
POINT_3D, PE6, XE6, YE1, ZE1;
POINT_3D, PGC4, XE5, YE1, ZE1-R42;
POINT_3D, PGC5, XE5, YE1+1, ZE1-R42;
POINT_3D, PGC6, XE5, YE1, ZE1+1-R42;
GENERALCYLINDER, GCL2, CNT1, L6, PGC4, PGC5, PGC6;
COLOUR, ' GREEN' ;
PRESENT, GCL2;
POINT_3D, PPA1, XE3-L12%22, YE1-R4+(W12%2-0.2*W12),
ZE1-R4-H12%2;

```

```

POINT_3D, PPA2, XE3- L12%22+L12, YE1- R4+ (W12%2- 0. 2*W12) ,
ZE1- R4- H12%2;
POINT_3D, PPA3, XE3- L12%22+L12, YE1- R4+ (W12- 0. 2*W12) ,
ZE1- R4;
PARALLELEPIPED, PAR1/CENLINPARA=(PPA1, PPA2, PPA3);
PRESENT, PAR1;
    POINT_3D, PPA4, XE3- L13%22, YE1+R4- (W13%2- 0. 2*W13) ,
        ZE1- R4- H13%2;
POINT_3D, PPA5, XE3- L13%22+L13, YE1+R4- (W13%2- 0. 2*W13) ,
ZE1- R4- H13%2;
POINT_3D, PPA6, XE3- L13%22+L13, YE1+R4+0. 2*W13,
ZE1- R4;
PARALLELEPIPED, PAR2/CENLINPARA=(PPA4, PPA5, PPA6);
PRESENT, PAR2;
    ASSIGN, XE7, L7+XE6;
    POINT_3D, PE7, XE7, YE1, ZE1;
    CYLINDER, CYL4, R7/COORDCYL=(PE6, PE7);
    COLOUR, ' CYAN' ;
    PRESENT, CYL4;
    ASSIGN, XE8, L8+XE7;
    POINT_3D, PE8, XE8, YE1, ZE1;
    CYLINDER, CYL5, R8/COORDCYL=(PE7, PE8);
    PRESENT, CYL5;
    ASSIGN, XE9, L9+XE8;
    POINT_3D, PE9, XE9, YE1, ZE1;
    CYLINDER, CYL6, R9/COORDCYL=(PE8, PE9);
    PRESENT, CYL6;
    ASSIGN, XE10, L10+XE9;
    POINT_3D, PE10, XE10, YE1, ZE1;
    CYLINDER, CYL7, R10/COORDCYL=(PE9, PE10);
    COLOUR, ' MAGENTA' ;
    PRESENT, CYL7;
    ASSIGN, XE11, A11+XE10;
    POINT_3D, PE11, XE11, YE1, ZE1;
    SPHERESEG, SPH1, R11/COORDSEG=(PE10, PE11);
    PRESENT, SPH1;
    ASSIGN, XE14, XE8- 170*L;
    ASSIGN, YS14, YE1- R8;
    ASSIGN, YE14, YS14- L14;
    POINT_3D, PS14, XE14, YS14, ZE1;
    POINT_3D, PE14, XE14, YE14, ZE1;
    CONE, CON2, R14B, R14A/COORDCONE=(PS14, PE14);
    COLOUR, ' YELLOW' ;
    PRESENT, CON2;
    ASSIGN, XE16, XE14;
    ASSIGN, YS16, YE14;
    ASSIGN, YE16, YE14- L14;
    POINT_3D, PS16, XE16, YS16, ZE1;
    POINT_3D, PE16, XE16, YE16, ZE1;
    CYLINDER, CYL8, R16/COORDCYL=(PS16, PE16);
    COLOUR, ' CYAN' ;
    PRESENT, CYL8;
    ASSIGN, XE18, XE16;
    ASSIGN, YS18, YE16;
    ASSIGN, YE18, YE16- L16;
    POINT_3D, PS18, XE18, YS18, ZE1;
    POINT_3D, PE18, XE18, YE18, ZE1;
    CYLINDER, CYL9, R18/COORDCYL=(PS18, PE18);
    PRESENT, CYL9;
    ASSIGN, COS1, COSD(ANG1);
    ASSIGN, SIN1, SIND(ANG1);
    ASSIGN, XS20, XE18;
    ASSIGN, XE20, XE18+SIN1*L20; ;
    ASSIGN, YS20, YE18;

```

```

ASSIGN, YE20, YE18-L20;
ASSIGN, ZE20, ZE1+COS1*L20;
ASSIGN, XM20, XS20+(1-COSD(45))*SIN1*L20;
ASSIGN, YM20, YE18-COSD(45)*L20;
ASSIGN, ZM20, ZE1+(1-COSD(45))*COS1*L20;
POINT_3D, PS20, XS20, YS20, ZE1;
POINT_3D, PM20, XM20, YM20, ZM20;
POINT_3D, PE20, XE20, YE20, ZE20;
TOROID, TOR1, R20/COORDTOR=(PS20, PM20, PE20);
PRESENT, TOR1;
ASSIGN, XS21, XE20;
ASSIGN, XE21, XE20+SIN1*L21;
ASSIGN, YS21, YE20;
ASSIGN, YE21, YE20;
ASSIGN, ZS21, ZE20;
ASSIGN, ZE21, ZS21+COS1*L21;
POINT_3D, PS21, XS21, YS21, ZS21;
POINT_3D, PE21, XE21, YE21, ZE21;
CYLINDER, CYL21, R21/COORDCYL=(PS21, PE21);
COLOUR, 'RED';
PRESENT, CYL21;
ASSIGN, XS22, XE21;
ASSIGN, XE22, XE21+SIN1*L22;
ASSIGN, YS22, YE21;
ASSIGN, YE22, YE21;
ASSIGN, ZS22, ZE21;
ASSIGN, ZE22, ZS22+COS1*L22;
POINT_3D, PS22, XS22, YS22, ZS22;
POINT_3D, PE22, XE22, YE22, ZE22;
CYLINDER, CYL22, R22/COORDCYL=(PS22, PE22);
PRESENT, CYL22;
ASSIGN, XS23, XE22;
ASSIGN, XE23, XE22+SIN1*L23;
ASSIGN, YS23, YE22;
ASSIGN, YE23, YE22;
ASSIGN, ZS23, ZE22;
ASSIGN, ZE23, ZS23+COS1*L23;
POINT_3D, PS23, XS23, YS23, ZS23;
POINT_3D, PE23, XE23, YE23, ZE23;
CONE, CON23, R23A, R23B/COORDCONE=(PS23, PE23);
COLOUR, 'WHITE';
PRESENT, CON23;
ASSIGN, XS24, XE23;
ASSIGN, XE24, XE23+SIN1*L24;
ASSIGN, YS24, YE23;
ASSIGN, YE24, YE23;
ASSIGN, ZS24, ZE23;
ASSIGN, ZE24, ZS24+COS1*L24;
POINT_3D, PS24, XS24, YS24, ZS24;
POINT_3D, PE24, XE24, YE24, ZE24;
CONE, CON24, R24A, R24B/COORDCONE=(PS24, PE24);
PRESENT, CON24;
ASSIGN, XS25, XE24;
ASSIGN, XE25, XE24+SIN1*L25;
ASSIGN, YS25, YE24;
ASSIGN, YE25, YE24;
ASSIGN, ZS25, ZE24;
ASSIGN, ZE25, ZS25+COS1*L25;
POINT_3D, PS25, XS25, YS25, ZS25;
POINT_3D, PE25, XE25, YE25, ZE25;
CYLINDER, CYL25, R25/COORDCYL=(PS25, PE25);
COLOUR, 'CYAN';
PRESENT, CYL25;
ASSIGN, XE15, XE14;

```

```

ASSIGN, YS15, YE1+R8;
ASSIGN, YE15, YS15+L15;
POINT_3D, PS15, XE15, YS15, ZE1;
POINT_3D, PE15, XE15, YE15, ZE1;
CONE, CON3, R15B, R15A/COORDCONE=(PS15, PE15);
COLOUR, 'YELLOW';
PRESENT, CON3;
ASSIGN, XE17, XE15;
ASSIGN, YS17, YE15;
ASSIGN, YE17, YE15+L17;
POINT_3D, PS17, XE17, YS17, ZE1;
POINT_3D, PE17, XE17, YE17, ZE1;
CYLINDER, CYL10, R17/COORDCYL=(PS17, PE17);
COLOUR, 'CYAN';
PRESENT, CYL10;
ASSIGN, XE19, XE17;
ASSIGN, YS19, YE17;
ASSIGN, YE19, YE17+L19;
POINT_3D, PS19, XE19, YS19, ZE1;
POINT_3D, PE19, XE19, YE19, ZE1;
CYLINDER, CYL11, R19/COORDCYL=(PS19, PE19);
PRESENT, CYL11;
ASSIGN, COS2, COSD(ANG2);
ASSIGN, SIN2, SIND(ANG2);
ASSIGN, XS26, XE19;
ASSIGN, XE26, XE19+SIN2*L26;
ASSIGN, YS26, YE19;
ASSIGN, YE26, YE19+L26;
ASSIGN, ZE26, ZE1+COS2*L26;
ASSIGN, XM26, XS26+(1-COSD(45))*SIN2*L26;
ASSIGN, YM26, YE19+COSD(45)*L26;
ASSIGN, ZM26, ZE1+(1-COSD(45))*COS2*L26;
POINT_3D, PS26, XS26, YS26, ZE1;
POINT_3D, PM26, XM26, YM26, ZM26;
POINT_3D, PE26, XE26, YE26, ZE26;
TOROID, TOR2, R26/COORDTOR=(PS26, PM26, PE26);
PRESENT, TOR2;
ASSIGN, XS27, XE26;
ASSIGN, XE27, XE26+SIN2*L27;
ASSIGN, YS27, YE26;
ASSIGN, YE27, YE26;
ASSIGN, ZS27, ZE26;
ASSIGN, ZE27, ZS27+COS2*L27;
POINT_3D, PS27, XS27, YS27, ZS27;
POINT_3D, PE27, XE27, YE27, ZE27;
CYLINDER, CYL27, R27/COORDCYL=(PS27, PE27);
COLOUR, 'RED';
PRESENT, CYL27;
ASSIGN, XS28, XE27;
ASSIGN, XE28, XE27+SIN2*L28;
ASSIGN, YS28, YE27;
ASSIGN, YE28, YE27;
ASSIGN, ZS28, ZE27;
ASSIGN, ZE28, ZS28+COS2*L28;
POINT_3D, PS28, XS28, YS28, ZS28;
POINT_3D, PE28, XE28, YE28, ZE28;
CYLINDER, CYL28, R28/COORDCYL=(PS28, PE28);
PRESENT, CYL28;
ASSIGN, XS29, XE28;
ASSIGN, XE29, XE28+SIN2*L29;
ASSIGN, YS29, YE28;
ASSIGN, YE29, YE28;
ASSIGN, ZS29, ZE28;
ASSIGN, ZE29, ZS29+COS2*L29;

```

```
POINT_3D, PS29, XS29, YS29, ZS29;
POINT_3D, PE29, XE29, YE29, ZE29;
CYLINDER, CYL29, R29/COORDCYL=(PS29, PE29);
COLOUR, ' WHITE' ;
PRESENT, CYL29;
ASSIGN, XS30, XE29;
ASSIGN, XE30, XE29+SIN2*L30;
ASSIGN, YS30, YE29;
ASSIGN, YE30, YE29;
ASSIGN, ZS30, ZE29;
ASSIGN, ZE30, ZS30+COS2*L30;
POINT_3D, PS30, XS30, YS30, ZS30;
POINT_3D, PE30, XE30, YE30, ZE30;
CYLINDER, CYL30, R30/COORDCYL=(PS30, PE30);
COLOUR, ' RED' ;
PRESENT, CYL30;
COLOUR, ' GREEN' ;
ENDMACRO;
```



## 17.4.4 The Possibility to Present a Text File on a Drawing

## Example 4

The macro LIST uses the TEXT and TEXTFILE statements to present a table on a drawing. The data are stored on a file. A heading is added and a frame is drawn.

The macro can be found by following this link:



[macro\\_list.txt](#)

The result of the macro is shown in the figure below.

DN	NAME	DIA	TH	QUA	MAT_NO
125	1	139.7	7.1	STEEL	276-15507
125	2	139.7	7.1	STEEL	276-15507
125	3	139.7	7.1	STEEL	276-15507
50	4	60.3	4.5	STEEL	274-15403
125	5	139.7	7.1	STEEL	276-15507
125	6	139.7	7.1	STEEL	276-15507
125	7	139.7	7.1	STEEL	276-15507
125	8	139.7	7.1	STEEL	276-15507
125	9	139.7	7.1	STEEL	276-15507
125	10	139.7	7.1	STEEL	274-15403
50	11	60.3	4.5	STEEL	274-15403
50	15	60.3	4.5	STEEL	274-15403
50	16	60.3	4.5		
125	19	139.7	7.1		

*Figure 17:3. The result from the execution of the macro LIST.*

```

MACRO, LIST;
  GET
    /POINT_2D=(' Give upper left corner of frame', FRM);
  ASSIGN, X, FRM/XCOORD;
  ASSIGN, Y, FRM/YCOORD;
  ASSIGN, TXH, 3.5;
  ASSIGN, ILSP, 1.5*TXH;
  ASSIGN, DX, 45*TXH;
  !
  ! Give number of lines
  !
  ASSIGN, N, 14;
  ASSIGN, DY, (N+3)*ILSP;
  !
  ! Create frame
  !
  POINT_2D, P1, X+DX, Y;
  POINT_2D, P2, X+DX, Y-DY;
  POINT_2D, P3, X, Y-DY;
  CONTOUR, CNT, FRM/LINEEND=P1/LINEEND=P2
    /LINEEND=P3/LINEEND=FRM;
PRESENT, CNT;
!
! Create vertical lines between the rows
!
POINT_2D, P4, X+6*TXH, Y;
POINT_2D, P5, X+6*TXH, Y-DY;
LINE, L1, P4/LINEEND=P5;
PRESENT, L1;
POINT_2D, P6, X+11*TXH, Y;
POINT_2D, P7, X+11*TXH, Y-DY;
LINE, L2, P6/LINEEND=P7;
PRESENT, L2;
POINT_2D, P8, X+19*TXH, Y;
POINT_2D, P9, X+19*TXH, Y-DY;
LINE, L3, P8/LINEEND=P9;
PRESENT, L3;
POINT_2D, P10, X+25*TXH, Y;
POINT_2D, P11, X+25*TXH, Y-DY;
LINE, L4, P10/LINEEND=P11;
PRESENT, L4;
POINT_2D, P12, X+33*TXH, Y;
POINT_2D, P13, X+33*TXH, Y-DY;
LINE, L5, P12/LINEEND=P13;
PRESENT, L5;
!
! Put a heading
!
ASSIGN, DY, ILSP;
POINT_2D, TXP, X, Y-DY;
TEXT, TXT, TXP/TEXTLINE=
'   DN   NAME   DIA   TH   QUA   MAT_NO' ;
PRESENT, TXT;
POINT_2D, P14, X, Y-DY-ILSP;
POINT_2D, P15, X+DX, Y-DY-ILSP;
LINE, L6, P14/LINEEND=P15;
PRESENT, L6;
!
! Put textfile
!
POINT_2D, TXP, X, Y-DY-2*ILSP;
TEXTFILE, TXF, 'STEEL.DAT' /POSLINES=(TXP, 1, N);
PRESENT, TXF;
ENDMACRO;

```

## 17.4.5 The Usage of Attributes in Geometry Macros

## Example 5

The macro ATTR shows the usage of the ATTRIBUTE statement. Note that it is possible to give the attribute number as an alias, a number or a variable.

```
MACRO,ATTR;
  ASSIGN,LS,105.6;
  ASSIGN,DS,0.2543;
  ATTRIBUTE,'CHAIR'/ATTRDATA=(I2,4)
/ATTRDATA=(R1,LS)
/ATTRDATA=(R2,DS)
/ATTRDATA=(R3,35.45)
/ATTRDATA=(S2,'I2: chair no')
/ATTRDATA=(S3,'R1-R3: chair data');
ATTRIBUTE,55/ATTRDATA=(I,3)
/ATTRDATA=(R4,1.234);
ASSIGN,ATTNO,7777;
GET/STR=('Give a string: ',STR);
ATTRIBUTE,ATTNO/ATTRDATA=('S1',STR);
ENDMACRO;
```

The result of a stand alone execution of the macro ATTR is shown below.

RUN SB\_SYSTEM:SZ006

Present Geometry Macro:

- (0) Exit
- (1) Print on terminal
- (2) Create 2D geometry and store on DB
- (3) Create 3D volume model and store on DB
- (4) Create 3D volume model + picture and store on DB

Enter activity : 1

General Component Data Bank not assigned!

Give name of macro to be run: ATTR

ATTNO = 1111

I1 0

I2 4

R1 105.60000

R2 0.25430

R3 35.45000

S1

S2 I2: chair no

S3 R1-R3: chair data

ATTNO = 55

R1 0.00000

R2 0.00000

R3 0.00000

R4 1.23400

S1

S2

```
S3      3
Give a string:
> Just a test string !!!
ATTNO =      7777
S1      Just a test string !!!
Once more ? N
Give name of macro to be run:
```

## 17.4.6 Layer Handling

## Example 6

The macro LAYER shows the layer alias and layer class facilities. Note that a layer can be given as an alias or as a number. The layer class can be given either as the class name or the class number, each preceded by '#'.  
 The result of a stand alone execution of the macro LAYER is shown below.

```
MACRO,LAYER;
  GET/STRING=('Layer alias : ',LALIAS)
  /STRING=('Layer class : ',LCLASS)
  /INTEGER=('Layer number: ',LNUMBER);
  LAYER,'NOLL';
  LAYER,'FORM';
  LAYER,1001;
  LAYER,'#DRA';
  LAYER,'#1';
  LAYER,LALIAS;
  LAYER,LCLASS;
  LAYER,LNUMBER;
ENDMACRO;
```

The result of a stand alone execution of the macro LAYER is shown below.

Present Geometry Macro:

- (0) Exit
- (1) Print on terminal
- (2) Create 2D geometry and store on DB
- (3) Create 3D volume model and store on DB
- (4) Create 3D volume model + picture and store on DB

```
Enter activity : 1
General Component Data Bank not assigned!
Give name of macro to be run: LAYER
```

Layer alias :

> FORM

Layer class :

> #AAA

Layer number:

> 12

LAYER & ALIAS:	0	ZERO
LAYER & ALIAS:	300	FORM
LAYER & ALIAS:	1001	VIEW1
LAYER CLASS:	10	DRA
LAYER CLASS:	1	PIPE
LAYER & ALIAS:	300	FORM
LAYER CLASS:	11	AAA
LAYER & ALIAS:	12	DOZEN

Once more ? Y

Layer alias :

> FROM

Layer class :



```
> #DRAW
Layer number:
> 100
LAYER & ALIAS:          0    ZERO
LAYER & ALIAS:          300   FORM
LAYER & ALIAS:          1001  VIEW1
LAYER CLASS:            10    DRA
LAYER CLASS:            1     PIPE
Layer alias FROM not defined. Layer ignored!
Layer class DRAW not found. Layer ignored!
LAYER:      100
Once more ? N
Give name of macro to be run:
```

## 17.4.7 The Facility of Using Conditional Statements in a Macro

## Example 7

This example illustrates the possibility of using the IF and WHILE statement to test the input data. Such a test can be useful to avoid runtime errors.

```
MACRO, INPUT;  
  ASSIGN, CONT, 1;  
  WHILE, CONT == 1;  
    GET/STR=('Key in colour', COL);  
    IF, COL == 'GREEN' OR COL == 'CYAN'  
      OR COL == 'BLUE' OR COL == 'MAGENTA'  
      OR COL == 'RED' OR COL == 'YELLOW'  
      OR COL == 'WHITE';  
    ASSIGN, CONT, 0;  
  ENDIF;  
ENDWHILE;  
!  
! When desired colour has been given the rest  
! of the macro is executed.  
!  
ENDMACRO;
```

It might be more convenient to have such a parameter check as a submacro which would look like follows.

```
MACRO, INPUT_COLOUR, COL;  
  DECLARE, COL, STRING;  
  ASSIGN, CONT, 1;  
  WHILE, CONT == 1;  
    GET/STR=('Key in colour', COL);  
    IF, COL == 'GREEN' OR COL == 'CYAN'  
      OR COL == 'BLUE' OR COL == 'MAGENTA'  
      OR COL == 'RED' OR COL == 'YELLOW'  
      OR COL == 'WHITE';  
    ASSIGN, CONT, 0;  
  ENDIF;  
ENDWHILE;  
ENDMACRO;
```

The usage of INPUT\_COLOUR could be like this.

```
MACRO, MAIN;  
  DECLARE, C, STRING;  
  CALL, INPUT_COLOUR, C;  
  COLOUR, C;  
  !  
ENDIF;
```

## 17.4.8 The Possibility to Write Recursive Geometry Macros

## Example 8

This example illustrates the possibility of writing a recursive macro, i.e. a macro which calls itself as a submacro. It uses a modified version of the macro RECTANGLE given in *Appendix 2*.

```
MACRO,RECPNT,PNT1,PNT2;
  DECLARE,PNT1,POINT_2D;
  DECLARE,PNT2,POINT_2D;
  CALL,RECTANGLE,PNT1,PNT2;
  ASSIGN,X1,PNT1/XCOORD;
  ASSIGN,Y1,PNT1/YCOORD;
  !
  ! When the condition is not fulfilled
  ! no further calls to RECPNT are made
  ! and the execution is terminated
  !
  IF,X1 < 250;
    ASSIGN,X2,PNT2/XCOORD;
    ASSIGN,Y2,PNT2/YCOORD;
    POINT_2D,PNT3,X2+X2-X1,Y2+Y2-Y1;
    CALL,RECPNT,PNT2,PNT3;
  ENDIF;
ENDMACRO;

MACRO,RECTANGLE,P1,P3;
  DECLARE,P1,POINT_2D;
  DECLARE,P3,POINT_2D;
  ASSIGN,X1,P1/XCOORD;
  ASSIGN,Y1,P1/YCOORD;
  ASSIGN,X3,P3/XCOORD;
  ASSIGN,Y3,P3/YCOORD;
  POINT_2D,P2,X1,Y3;
  POINT_2D,P4,X3,Y1;
  CONTOUR,CNT,P1
    /LINEEND = P2
    /LINEEND = P3
    /LINEEND = P4
    /LINEEND = P1;
  PRESENT,CNT;
ENDMACRO;
```

## 17.4.9 Change Colour in Drawing by Using Geometry Macro

## Example 9

This macro is supposed to be started from the General Diagrams application.

All cables without interference class and component defined will be drawn red.

All cables with both interference class and component defined will be drawn green.

All cables with either interference class or component defined will be drawn blue.

The same function for Equipments with the data room and component.

```
MACRO, OUTF_DIAG_COLOUR;
! Declarations
DECLARE, A1,      STRING;
DECLARE, COMPNAME,    STRING;
DECLARE, DELIM,      STRING;
DECLARE, DWG,      STRING;
DECLARE, E1,      EXTRACT;
DECLARE, INDEX,    INTEGER;
DECLARE, INT_C,      STRING;
DECLARE, LOOPMAX,    INTEGER;
DECLARE, MODNAME,    STRING;
DECLARE, PROJ,      STRING;
DECLARE, ROOM,      STRING;
DECLARE, STAT,      INTEGER;
DECLARE, SUB1,      STRING;
DECLARE, SUB2,      STRING;
DECLARE, SUB3,      STRING;

! Initiate
ASSIGN, DELIM, '-';
! Get the drawing name
DRAWING_NAME, DWG;
! Use data extraction to see which cables that are drawn in the diagram
ASSIGN, A1, 'DRA(DWG).VIE(*) .NCABLE';
EXTRACT, E1, A1;
GET/EXTRACT=(LOOPMAX, STAT, E1, DWG, , );
IF, STAT == 1;
LOOP, INDEX, 1:LOOPMAX;
ASSIGN, A1, 'DRA(DWG).VIE(*) .CABLE(INDEX).NAME';
EXTRACT, E1, A1;
GET/EXTRACT=(MODNAME, STAT, E1, DWG, , INDEX, );
IF, STAT == 1;
! Get the cable data by using data extraction
SPLIT, MODNAME, DELIM, PROJ, CAB_SYS, CAB_NAM, SUB2, SUB3;
ASSIGN, CABLE_NAME, CAB_SYS&DELIM&CAB_NAM;
ASSIGN, A1, 'CABLE(PROJ).CAB_M(CABLE_NAME). COMP_N'; EXTRACT, E1, A1;
GET/EXTRACT=(COMPNAME, STAT, E1, PROJ,
CABLE_NAME, );
IF, STAT == 0;
```

```

ASSIGN,COMPNAME,'';
ENDIF;
ASSIGN,A1,
'CABLE(PROJ).CAB_M(CABLE_NAME).EL_PROP.INT_C';
EXTRACT,E1,A1;
GET/EXTRACT=(INT_C,STAT,E1,PROJ,CABLE_NAME,,);
IF,STAT == 0;
ASSIGN,INT_C,'';
ENDIF;
! Draw cables without any data red
IF,COMPNAME == '' AND INT_C == '';
CHANGEDRAW,'CABLE',MODNAME
/MARKINGCOLOUR='RED';
ELSE;
! Draw cables with all data green
IF,COMPNAME /= '' AND INT_C /= '';
CHANGEDRAW,'CABLE',MODNAME
/MARKINGCOLOUR='GREEN';
! Draw cables with some data blue
ELSE;
CHANGEDRAW,'CABLE',MODNAME
/MARKINGCOLOUR='BLUE';
ENDIF;
ENDIF;
ENDIF;
ENDLOOP;
ENDIF,
! Use data extraction to see which equipments that are drawn in the diagram
ASSIGN,A1,'DRA(DWG).VIE(*) .NEQUIP';
EXTRACT,E1,A1;
GET/EXTRACT=(LOOPMAX,STAT,E1,DWG,,);
IF,STAT == 1;
LOOP,INDEX,1:LOOPMAX;
ASSIGN,A1,'DRA(DWG).VIE(*) .EQUIP(INDEX).NAME';
EXTRACT,E1,A1;
GET/EXTRACT=(MODNAME,STAT,E1,DWG,,INDEX,);
IF,STAT == 1;
! Get the equipment data by using data extraction
SPLIT,MODNAME,DELIM,PROJ,EQ_NAME,SUB1,SUB2,SUB3;
ASSIGN,A1,'EQUIP(PROJ).ITEM(EQ_NAME).COMP_N';
EXTRACT,E1,A1;
GET/EXTRACT=(COMPNAME,STAT,E1,PROJ,EQ_NAME,);
IF,STAT == 0;
ASSIGN,COMPNAME,'';
ENDIF;
ASSIGN,A1,'EQUIP(PROJ).ITEM(EQ_NAME).ROOM';
EXTRACT,E1,A1;
GET/EXTRACT=(ROOM,STAT,E1,PROJ,EQ_NAME,);

```



```

IF,STAT == 0;
ASSIGN,ROOM,'';
ENDIF;
! Draw equipment items without any data red
IF,COMPNAME == '' AND ROOM == '';
CHANGEDRAW,'EQUIP',MODNAME/
MARKINGCOLOUR='RED';
ELSE;
! Draw equipment items with all data green
IF,COMPNAME /= '' AND ROOM /= '';
CHANGEDRAW,'EQUIP',MODNAME
/MARKINGCOLOUR='GREEN';
! Draw equipment items with some data blue
ELSE;
CHANGEDRAW,'EQUIP',MODNAME
/MARKINGCOLOUR='BLUE';
ENDIF;
ENDIF;
ENDIF;
ENDLOOP;
ENDIF,
ENDMACRO;

```

The following macro would change the colours of the equipments and cables back to be the one defined by the default file.

```

MACRO, OUTF_DIAG_DEFAULT;
! Declarations
DECLARE, A1,      STRING;
DECLARE, DWG,     STRING;
DECLARE, E1,      EXTRACT;
DECLARE, INDEX,   INTEGER;
DECLARE, LOOPMAX, INTEGER;
DECLARE, MODNAME, STRING;
DECLARE, STAT,    INTEGER;
! Get the drawing name
DRAWING_NAME,DWG;
! Use data extraction to see which cables that are drawn in the diagram
ASSIGN,A1,'DRA(DWG).VIE(*) .NCABLE';
EXTRACT,E1,A1;
GET/EXTRACT=(LOOPMAX,STAT,E1,DWG,,);
IF,STAT == 1;
LOOP,INDEX,1:LOOPMAX;
ASSIGN,A1,'DRA(DWG).VIE(*) .CABLE(INDEX).NAME';
EXTRACT,E1,A1;
GET/EXTRACT=(MODNAME,STAT,E1,DWG,,INDEX,);
IF,STAT == 1;
CHANGEDRAW,'CABLE',MODNAME
/MARKINGCOLOUR='DEFAULT';
ENDIF;

```

```

ENDLOOP;
ENDIF;
! Use data extraction to see which equipments that are drawn in the diagram
ASSIGN,A1,'DRA(DWG).VIE(*).NEQUIP';
EXTRACT,E1,A1;
GET/EXTRACT=(LOOPMAX,STAT,E1,DWG,,);
IF,STAT == 1;
LOOP,INDEX,1:LOOPMAX;
ASSIGN,A1,'DRA(DWG).VIE(*).EQUIP(INDEX).NAME';
EXTRACT,E1,A1;
GET/EXTRACT=(MODNAME,STAT,E1,DWG,,INDEX,);
IF,STAT == 1;
CHANGEDRAW,'EQUIP',MODNAME
/MARKINGCOLOUR='DEFAULT';
ENDIF;
ENDLOOP;
ENDIF;
ENDMACRO;

```